A Guide to Specifying Observation Equations for the Estimation of DSGE Models

Johannes Pfeifer*

Universität der Bundeswehr München

October 11, 2021

First version: July 2013

Abstract

This paper provides an introduction to specifying observation equations for estimating DSGE models. It covers both linearized and non-linearized DSGE models. While Dynare 4 is used to illustrate the actual computer implementation, the principles outlined apply in general.

Keywords: Dynare 4, observation equations, state space representation, DSGE, estimation.

*Special thanks go to Benjamin Born for helpful suggestions. I also want to thank Roland von Campe, Xin Gu, Jochen Güntner, Beka Lamazoshvili, Ruthira Naraidoo, Tri Dung Nguyen, and Francesco Turino. All errors are my own. For corrections or suggestions, please email me at johannes.pfeifer@unibw.de.

Contents

List of Figures

Listings

List of Abbreviations

AR	Autoregressive.
BEA	Bureau of Economic Analysis.
BGP	Balanced Growth Path.
CME	Classical Monetary Economy.
CPI	Consumer Price Index.
CPS	Current Population Survey.
DSGE	Dynamic Stochastic General Equilbrium.
GDP	Gross Domestic Product.
HP	Hodrick-Prescott.
ILO	International Labour Organization.
IRF	Impulse Response Function.
MCMC	Monte Carlo Markov Chain.
MFP	Multifactor Productivity.
NIPA	National Income and Product Accounts.
R&D	Research and Development.
RBC	Real Business Cycle.
SNA	System of National Accounts 2008.
TFP	Total Factor Productivity.
VAR	Vector Autoregression.

1 Introduction

This is a guide to specifying equations for estimating DSGE models. While the central focus is on the specification of observation equations, I will also provide some hands-on advice more generally related to Dynare 4 and estimation in general. I put this advice in *Remark boxes*. Upon first reading this document, I recommend to simply ignore those remarks and focus on the big picture. Once you delve into the actual implementation, going back to the remarks will be helpful.

The corresponding mod-files and the data used in this document can be downloaded at https://sites.google.com/site/pfeiferecon/dynare. All data are taken from the FRED database provided by the Federal Reserve Bank St. Louis and are available at http://research.stlouisfed.org/fred2/. Data mnemonics are provided in brackets.

1.1 A Baseline RBC Model

For our purpose, it's useful to lay down a simple one-good Real Business Cycle (RBC) model of the Classical Monetary Economy (CME) type. We assume that labor does not provide any disutility. Thus, labor hours h_t will be fixed at their maximum, i.e. $h_t = 1$. We will abstract from growth in total population N_t (in which case total hours in the economy would be $N_t h_t$) by assuming the model to be in per capita terms. The central planner problem is given by:

$$\max_{\{C_t,I_t,K_t,B_t\}_{t=0}^{\infty}} E_0 \sum_{t=0}^{\infty} \beta^t \log(C_t)$$

subject to

$$C_t + I_t + \frac{B_t}{P_t} = Y_t + \frac{B_{t-1}R_{t-1}}{P_t} \tag{1}$$

$$Y_t = A_t K_{t-1}^{\alpha} \left(X_t h_t\right)^{1-\alpha} = A_t K_{t-1}^{\alpha} X_t^{1-\alpha} \tag{2}$$

$$K_t = (1-\delta)K_{t-1} + I_t \tag{3}$$

$$\frac{R_t}{\bar{R}} = \left(\frac{\Pi_t}{\bar{\Pi}}\right)^{\phi} \tag{4}$$

$$\text{some law of motion for } A_t \text{ and } X_t \tag{5}$$

given $K_{-1} > 0$ and suitable transversality conditions

Here E_t denotes the expectations operator conditional on time t information, C_t is consumption, I_t investment, K_t is capital, B_t a private nominal bond in zero net supply,[1] P_t is the price of the one good in this economy, A_t is Total Factor Productivity (TFP), X_t is labor augmenting

[1]This implies that $B_t = 0 \,\forall\, t$.

technology growth, $\Pi_t = P_t/P_{t-1}$ denotes inflation, and R_t is the gross nominal interest rate. The household has log-utility and discounts with discount factor β. Equation (1) is the economy's resource constraint, while equation (2) is a Cobb-Douglas aggregate production function with capital share α. Equation (3) is the law of motion for capital with geometric depreciation rate δ and equation (4) specifies the conduct of monetary policy, which is assumed to follow a Taylor type rule with inflation feedback parameter $\phi > 1$, where $\bar{R}$ and $\bar{\Pi}$ represent the steady state level of the nominal interest and inflation, respectively. As is well-known (see e.g. Galí 2008, Chapter 2), due to the absence of nominal rigidities, this economy exhibits the classical dichotomy. Hence, while keeping the model simple due to the real part just being the standard RBC mode, the model setup also allows studying the case of nominal variables as it would appear in standard New Keynesian models.

Remark 1 (Timing convention)
We use the *end of period stock* timing convention of Dynare for predetermined endogenous variables like capital. That is, variables get the timing at which they are determined. For example, the capital stock used at time t in the production function for Y_t was determined by investment at time $t-1$. Thus, the production function is $Y_t = K_{t-1}^{\alpha} X_t^{1-\alpha}$ and the law of motion for capital $K_t = (1-\delta)K_{t-1} + I_t$. Hence, the timing of capital is shifted by one period compared to the more common *beginning of period stock* notation typically employed in papers. If you want to use the beginning of period stock timing convention in Dynare, you need to use the `predetermined_variables`-command. Note, however, that even when using this command, Dynare will still internally use the end of period stock timing convention when plotting Impulse Response Functions (IRFs) or computing smoothed variables (see the manual).

For the moment, we keep the joint technology process $A_t X_t^{1-\alpha}$ generic in order to encompass both stationary technology processes and non-stationary processes containing deterministic or stochastic trends. Thus, variables like C_t and K_t in this generic model can represent either stationary concepts or trending concepts, depending on the underlying technology process. We will revisit this issue later. After substituting for investment and output, the well-known first order conditions to this problem are (ignoring non-negativity and transversality conditions):

$$\frac{1}{C_t} = \beta E_t \frac{1}{C_{t+1}} \left[\alpha A_{t+1} K_t^{\alpha-1} X_{t+1}^{1-\alpha} + (1-\delta)\right] \tag{6}$$

$$\frac{1}{C_t P_t} = \beta E_t \frac{1}{C_{t+1}} \frac{R_t}{P_{t+1}} \tag{7}$$

$$C_t + K_t - (1-\delta)K_{t-1} + \frac{B_t}{P_t} = A_t K_{t-1}^{\alpha} X_t^{1-\alpha} + \frac{B_{t-1} R_{t-1}}{P_t} \tag{8}$$

$$\text{some law of motion for } A_t \text{ and } X_t \tag{9}$$

$$\frac{R_t}{\bar{R}} = \left(\frac{\Pi_t}{\bar{\Pi}}\right)^{\phi} \tag{10}$$

Unfortunately, as in New Keynesian models, the nominal price level P_t is not uniquely determined, only relative prices and inflation Π_t are.[2] Thus, the equations involving prices (and nominal variables in general), (7) and (8), have to be rewritten so that they only involve inflation and real variables before entering them into the computer:

$$\frac{1}{C_t} = \beta E_t \frac{1}{C_{t+1}} \frac{R_t}{\Pi_{t+1}} \tag{11}$$

$$C_t + K_t - (1-\delta)K_{t-1} + \frac{B_t}{P_t} = A_t K_{t-1}^{\alpha} X_t^{1-\alpha} + \frac{B_{t-1}}{P_{t-1}} R_{t-1} \frac{1}{\Pi_t} \tag{12}$$

where B_t/P_t now denotes real bonds.

From now on, we are going to impose the market clearing condition that private bonds are in zero net supply on the budget constraint ,[3] i.e. $B_t/P_t = 0 \,\forall\, t$. Hence,

$$C_t + K_t - (1-\delta)K_{t-1} = A_t K_{t-1}^{\alpha} X_t^{1-\alpha} \tag{13}$$

There are models where this is not the case, e.g. if bonds are government bonds that are not in zero net supply. In this case, B_t/P_t would be entered as one single variable into Dynare.

For the beginning we are also going to abstract from growth in labor-augmenting technology X_t by assuming $X_t = 1 \,\forall\, t$. The reason for this will become clear shortly. Moreover, in order to get a mod-file that can actually be run, we need to assume some functional form for A_t. It is often convenient to work with a log-normal $AR(1)$-process:

$$A_t = e^{z_t} \tag{14}$$

[2] The nominal price level P_t is special compared to all other variables considered in this document. While it is not stationary (unless we assume a money demand and supply process that pins it down), it is not a trending variable in the sense that it does not grow with all other variables along a Balanced Growth Path (BGP). Thus, it will get a special treatment in that we are not going to detrend it and generally keep it in uppercase letters indicating undetrended variables.

[3] Note that you can only impose this market clearing condition after taking the first order conditions. It would be invalid to eliminate bonds already in the budget constraint of the household. Even if bonds are in zero net supply, household savings behavior in equilibrium still needs to be consistent with the bond market clearing.

$$z_t = \rho z_{t-1} + \varepsilon_t,\ \varepsilon_t \sim \mathcal{N}(0, \sigma^2) \tag{15}$$

The resulting nonlinear mod-file that incorporates the market clearing condition for bonds and abstracts from growth in labor-augmenting technology X_t is given by Listing 1. It also introduces a plausible parametrization.

Listing 1: Basic RBC Classical Monetary Economy Model

```
var y c k A z R Pi;
varexo eps_z;

parameters beta delta alpha rhoz phi_pi Pibar;

alpha   = 0.33;    // capital share
delta   = 0.025;   //deprecation rate
beta    = 0.99;    //discount factor
Pibar   = 1;
phi_pi  = 1.5;
rhoz    = 0.97;    //TFP autocorr. from linearly detrended Solow residual

model;
#Rbar    = 1/beta;
1/c=beta*1/c(+1)*(alpha*A(+1)*k^(alpha-1)+(1-delta));
1/c=beta*1/c(+1)*(R/Pi(+1));
A*k(-1)^alpha=c+k-(1-delta)*k(-1);
y=A*k(-1)^alpha;
R/Rbar=(Pi/Pibar)^phi_pi;
A=exp(z);
z=rhoz*z(-1)+eps_z;
end;

steady_state_model;
k=((1/beta-(1-delta))/alpha)^(1/(alpha-1));
y=k^alpha;
c=y-delta*k;
R=1/beta;
Pi=Pibar;
A=1;
z=0;
end;

shocks;
var eps_z=0.0068^2; //estimated value
end;
```

```
steady;
check;

stoch_simul(order=1,irf=20,periods=250);
```

It will become clear shortly why labor augmenting technology growth X_t is not considered in the mod-file and why we use lowercase letters to represent the model variables. We will often refer to these equations.

Remark 2 (Using stoch_simul before Estimation)
You can see that Listing 1 first explicitly initializes all parameters and then uses the sequence

steady;

check;

stoch_simul;

You might think: what is the point in fully calibrating the model to some rather arbitrary parameter values? I am dealing with observation equations because I want to estimate (a subset of) those parameters. And you are right. When you want to estimate your model, none of this is actually required. It would be sufficient to just specify the values of the parameters not to be estimated. However, fully calibrating your model and using the sequence of commands above is *strongly recommended* when you are still tweaking around with your model as it serves as a useful cross-check: can the steady state be correctly computed for the calibration? Are the Blanchard-Kahn conditions satisfied? Do the IRFs look sensible?

If you cannot get your model to run when you get to pick the parameters, estimation typically won't help either! Moreover, a fully calibrated model allows using the `use_calibration` option of the `estimated_params_init`-block to easily start the estimation at parameter values that you think are likely.

When everything is up and running, you can still uncomment the three lines above and directly go to estimation.

Remark 3 (Variable Naming in Matlab and Dynare)
There is a convention in economics/mathematics to assign Greek letters to parameters. This is unproblematic for writing down equations in papers, but can cause serious problems when working with programming languages. The reason is that this creates a huge potential for naming conflicts. Listing 1 for example uses `alpha` to denote the capital share. But `alpha`

is also the name of the Matlab function for setting the transparency in figures. Similarly, `beta` is both the name for the discount factor and the Matlab command to call the beta function. When working with pure Dynare as I do in this document, this is not an issue. Hence, for better readability, I will ignore my own warning in this document. But you should do as I say, not as I do, because problems typically start to arise as soon as you try to interface Dynare with your own Matlab code, e.g. using a steady state file. Thus, generally it is *strongly discouraged* to use correctly spelled Greek letters. Rather, use something like `alppha` and `betta`. Similarly, try avoiding to name investment simply `i` as this is also the imaginary number. Better name it `invest`. Finally, you must never use Dynare commands or options as variable or parameter names. For example, do not use `sigma_e` or `Ln`. If in doubt, consult the manual (Adjemian et al. 2011). Heeding these warnings may save you a lot of trouble down the road.

Remark 4 (Parameter dependence and the use of model-local variables)
You might have recognized that the steady state interest rate `Rbar` has been defined as a model-local variable using the #-operator in the `model`-block of Listing 1. This means that `Rbar` is not an independent parameter, but Dynare substitutes `1/beta` whenever it encounters `Rbar`. If one only wants to solve and simulate the mod-file using the `stoch_simul`-command, one could have defined `Rbar` as an additional parameter. But for estimation, this could have disastrous consequences! `Rbar` is not a true independent parameter, but has a one-to-one relationship with the value of `beta`: whenever `beta` changes, so should `Rbar`. But parameters explicitly specified in the `params`-statement and not subsequently estimated are initialized once before the run of `estimation` and are not updated. This is no problem if `beta` is not estimated and `Rbar` is thus constant at the fixed value of `1/beta` (1/0.99 in the above example). However, if one estimates `beta` and does define `Rbar` as an independent parameter, `beta` will be changed during estimation but `Rbar` will not. Let's say the Monte Carlo Markov Chain (MCMC) tries the value $\beta = 0.98$. Then $\bar{R}$ will be kept fixed at the initial value of `1/beta` before estimation started, i.e. 1/0.99.

Now you might want to think about then also adding `Rbar` to the estimated parameters, but this is also not feasible. There is a one-to-one relationship between `Rbar` and `beta` that needs to be obeyed. You cannot simply treat both as independent parameters and estimate them, because then the MCMC might try 0.98 for `beta` and 1/0.97 for `Rbar`.
Thus, never define a function of estimated parameters as an independent parameter! The parameter will not be correctly updated, your estimation results will be wrong, and you won't

even get an error message!

1.2 State-Space Setup

For shaping our thinking, it is useful to think about the purpose of solving a model for a moment. The goal is to find recursive policy functions for the endogenous state and control variables that express them as a function of only the endogenous and exogenous states (and the parameters). Such a solution of Dynamic Stochastic General Equilbrium (DSGE) models can typically be written in basic state-space form as

$$x_t = g\left(x_{t-1}, \varepsilon_t^{struct}\right) \tag{16}$$

$$y_t^{obs} = h\left(x_t, \varepsilon_t^{obs}\right) \;, \tag{17}$$

where x_t are the state variables, ε_t^{struct} are the structural shocks, y_t^{obs} are the observed variables[4] and ε_t^{obs} is measurement error. The function g is the policy function for the states, while h provides the policy function for the observables (which are a subset of the controls). Equation (16) is called the *state-transition equation*, which describes the evolution of the state variables x_t given the past state of the system and the realization of the structural shocks, ε_t^{struct}. Equation (17), the *observation equation*, describes how the observed variables map into the state variables, potentially including measurement error ε_t^{obs}. In case of no measurement error, the covariance matrix of ε_t^{obs} is just a 0 matrix. For a linearized model, the policy functions g and h are simply linear functions.

1.3 Dynare's Treatment of Observation Equations

In order to estimate a DSGE model in Dynare, one has to specify the observation equation. Listing 2 provides the stylized Dynare code that is required to implement the following steps. Fortunately, specifying observation equations must only be done indirectly as Dynare will compute the mapping from state variables x_t into observables y_t^{obs} for us, provided we tell it how the observed data is related to the other variables in the model. But this also means that unless our observed variables y_t^{obs} exactly correspond to an actual model variable, we will have to add separate equations detailing how y_t^{obs} is linked to the variables of the model. After specifying these equations, the observables are simply treated as any other endogenous model

[4] In the state-space representation, x and y are used as generic variables. In later paragraphs, x and y will be used to denote specific economic concepts like technology and output as encountered in the introductory model.



Figure 1: Simulated output series from the mod-file presented in Listing 1.

variable. Computing the model solution will automatically deliver the required mapping. This is also the reason, why you have to add the observed variable y^{obs} defined in the observation equation to the endogenous model variables using the `var`-statement of the mod-file.

The most crucial issue when specifying observation equations is that the model variables like output y_t are typically stationary (see Figure 1 for an output series generated from a simple model), while empirically observed Gross Domestic Product (GDP) is non-stationary due to the presence of a growth trend in X_t (see Figure 2). This issue will be discussed in more detail in Section 1.4.

Given a specified observation equation, the next thing you have to do is providing Dynare with information which model variables are the observed variables. This is done using the `varobs`-command. Finally, you have to provide Dynare with a datafile that contains the empirical data series corresponding to the observed variable.

Remark 5 (Naming of variables in the datafile)
The variables in the datafile specified in the `datafile`-command of the mod-file must have the same names as the variables specified in the `varobs`-command for Dynare being able to link them to the model. For Excel files this implies using a corresponding column header, while when using a Matlab-file the variables must be named correctly. Note that Dynare is case-sensitive. So if your variable in the `varobs`-command is a lowercase y and your datafile



Figure 2: The solid line depicts US real GDP in constant 2005 dollars (FRED mnemonic gdpc96).

only has an UPPERCASE variable Y, you will get an error.

Summarizing, you have to follow four steps:

1. Add any observed variable y^{obs} that is not already a model variable to the endogenous model variables in the `var`-statement.
2. Add any required explicit observation equations to the `model`-block.
3. Tell Dynare which variables are observed using the `varobs`-command.
4. Provide a valid datafile with the correct naming of variables/columns and pass it to the `estimation`-command with the `datafile`-option.

Remark 6 (Warning Regarding Datafile Names)
Never give your datafile the name "data" or the same name as your mod-file! It will result in naming conflicts.

Listing 2: Basic structure of specifying observation equations

```
var y, Pi, R, mu, ... y_obs Pi_obs R_obs; //endogenous variables including observed variables
var_exo ... eps_y_obs; //exogenous shocks including explicitly specified measurement error
model;
...
Pi_obs=log(Pi); //observation equations linking model variables to observed variables
R_obs=R;
y_obs=log(y)-log(y(-1))+mu+eps_y_obs;
end;

varobs y_obs R_obs Pi_obs; //specifies which variables are observed
estimation(...,datafile=mydatafile); //defines the name of the datafile that includes variables named y_obs, R_obs, and Pi_obs
```

1.4 The Need for Specifying Observation Equations

The fundamental issue is that the observed data usually does not exactly match the model variables. The most common reason is the presence of a (stochastic) trend in the data, i.e. there is labor augmenting technology growth X_t.[5] In the following, the exposition will often move along the example of a growth trend, but the general techniques are more widely applicable. The reason for model variables being stationary in Dynare is that models are solved using perturbation techniques, i.e. using a Taylor approximation around an approximation point, typically the deterministic steady state. This requires the model to actually have a well-defined steady state. This was also the reason why we abstracted from labor augmenting technology growth X_t when we entered the introductory model into Dynare in Listing 1.

However, actual economies don't tend to converge back to a steady state, but grow over time. Typically, economists conceptualize this as the economy having a "steady state" in intensive form (per technology-weighted capita) and modeling the intensive form variables. The actual (per capita) variables are then growing along a BGP, i.e. output, consumption, and investment per capita are growing at the rate of technology growth X_t, while total output, consumption, and investment are growing at the rate of population and technology growth.

There are two general ways to get around the problem of modeling an economy in intensive form and only observing actual growing variables:

[5] The reason for representing technology as labor augmenting is that it guarantees the existence of a BGP. (See e.g. Acemoglu 2008).

1. Completely abstract from growth and write down a stationary model in intensive form that only describes the behavior of the economy around the BGP, while abstracting from the movement along the BGP itself (e.g. Kliem and Kriwoluzky 2014). We effectively encountered such a case in Listing 1.

⇒ Enter data made stationary/transformed into intensive form (e.g. using one-sided HP-filter, linear trend, etc.). This will be the topic of Section 4.

2. Write down a non-stationary model that models growth explicitly and then detrend the model equations to get a stationary model in intensive form with i) a well-defined steady state and ii) a model-consistent description as to how the BGP evolves (see e.g. Smets and Wouters 2007, for an example). This will be the topic of Section 5.

⇒ Enter data in the form implied by the theory of the model

We will see examples of both cases in the following.

2 Conventions

For trending variables, uppercase letters will denote trending variables, while lowercase letters will denote the corresponding intensive form/stationary variables. Thus, Y_t is aggregate GDP, while y_t is detrended GDP. Non-trending variables like the interest rate R_t will always use uppercase letters. Empirical data will be denoted with the superscript *data* so that Y_t^{data} is observed GDP as plotted in Figure 2. When referring to model variables, we have to distinguish between the model entered into Dynare's `model`-block, which usually must be entered in stationary form and the actual economic model as written down in papers. Sometimes both model concepts coincide (see e.g. the textbook treatment of the New Keynesian model in Galí 2008), but at other times the economic model needs to be transformed in order to be entered into Dynare. You encountered a first example in the initial model where nominal Bonds B_t occur in the description of the model, but only real bonds B_t/P_t could be entered into Dynare. Unless stated otherwise, when I refer to model variables I refer to the Dynare model.

The transformation of the empirical data that is matched to the model variables will be denoted with the superscript *obs*. For example, y_t^{obs} will denote the GDP data series specified in the `varobs`-command specifying which output measure is observed. Thus, y_t^{obs} is both the name of a series in the data-file given to Dynare using the `datafile`-option of the `estimation`-command and the name of a model variable specified in the `vars`-command at the beginning of the mod-file (see also Listing 2).

Finally, hats denote variables in percentage deviations from steady state, tildes denote the logarithm of a variable, and bars indicate steady state values. Hence, $\hat{y}_t$ denotes deviations of intensive form output from its steady state $\bar{y}$ and $\tilde{y}_t$ is the logarithm of output in intensive form.

Moreover, we will assume a model at quarterly frequency. We will see that this typically makes an adjustment of the interest rates necessary as they are usually quoted as annualized interest rates and not as quarterly interest rates. To summarize:

- Y_t: Aggregate output (still trending due to labor augmenting technology growth)
- y_t: Output in intensive form
- Y_t^{data}: Empirically observed output data used in data transformations
- y_t^{obs}: Final data transformation used in both the Dynare model and the datafile to link empirical data and model variables
- $\bar{y}$: Steady state of output in intensive form
- $\tilde{y}_t$: Logarithm of output in intensive form
- $\hat{y}_t$: Percentage deviation of output in intensive form from its steady state

3 Data Preliminaries

In the following, we will focus on three observed variables that we want to match: real output, inflation, and the nominal interest rate. These are three interesting cases, because

- Output is a trending variable. Hence, its treatment is the same as for other variables like consumption, investment or government spending. Due to the trend, we must somehow specify a mapping between the non-stationary data, Y_t^{data}, and the stationary output concept used in the model, y_t or $\hat{y}_t$.
- Inflation is a stationary variable which has a direct equivalent in economic models.
- The nominal interest is a stationary variable, but in contrast to inflation it is typically measured at a different frequency in the data than in the model.

The following subsections show some generic issues of data treatment that generally arise independently of the actual type of model considered. Rather, they relate to general data transformations needed to bring empirical data closer to economic model concepts.



Figure 3: Top panel: Logarithm of Federal government current tax receipts, Not Seasonally Adjusted (FRED: W006RU1Q027NBEA). Bottom panel: quarterly growth rate of Federal government current tax receipts, Not Seasonally Adjusted.

3.1 Two Warnings

Before proceeding, two warnings are in order that should be heeded when estimating DSGE models.

3.1.1 Seasonal Patterns

DSGE models are typically built to capture variations at business cycle frequency. They are ill-equipped to deal with variation at seasonal frequency. Thus, you should make sure that all your data are seasonally adjusted.[6] For the US, this is typically not an issue as the Bureau of Economic Analysis (BEA) mostly provides seasonally adjusted series. However, in other countries this is often not the case and it is essential to check. A seasonal pattern can often be easily identified by visually inspecting the quarterly growth rates. Figure 3 shows the US Federal government current tax receipts, which are not seasonally adjusted. Here, a clear seasonal pattern introduced by tax due dates is visible and needs to be removed, before the series can be used in estimation (unless you model this pattern explicitly).[7]

[6] See Granger (1979) for a justification of this approach.

[7] Seasonal adjustment is typically done using either a variant of X11 (Shiskin, Young, and Musgrave 1965) or Tramo-Seats (Gómez and Agustín Maravall 1996; Agustń Maravall 1999). Recently, both procedures have been made easily available at https://www.census.gov/data/software/x13as.html, see also U.S. Census Bureau (2013).

3.1.2 Data Revisions

There are two types of data series that differ according to how they are updated once important new information becomes available. There are

- Data where the whole series is regularly revised/recontrolled. For example, the US BEA conducts "flexible annual revisions" and "flexible comprehensive revisions" that revise the statistics for the whole time period to reflect updated source data and changes in definitions and methods (for more information, see Kornfeld et al. 2008). Starting in June 31st, 2013 the US National Income and Product Accounts (NIPA) tables incorporate the capitalization of Research and Development (R&D) expenditures advocated in the System of National Accounts 2008 (SNA). As a result, R&D expenditures that were previously treated as inputs are now treated as investment. This change was applied retroactively and significantly changed GDP figures for e.g. the 1970s. This new GDP series supersedes the old ones.

- Data that are collected on a "best level" basis, i.e. the data at any point in time t reflect the best estimate for the level of this series, given all information that was available up to time t. Any additional information becoming available at time $t+1$ is only used on a "forward basis", i.e. to estimate data points for time $t+i, i \geq 1$, but not for updating previous time points.

There are different problems associated with the two types of data. The first type of data, which is regularly updated for the whole time period, provides our best estimate of the state of the economy given currently available information. Thus, estimation in most applications uses the most recent vintage, because this vintage should provide the most reliable data. However, there are other applications where using the most recent data vintage is not advocated, because it does not reflect the information set agents had at the time they had to make their decisions. For example, if you want to study whether investors were rational in judging the Greek risk of default before the European sovereign debt crisis started in 2010, you should most probably look at real-time data, i.e. the GDP information people had at the time they made their investment decisions. In contrast, the most recent data vintage already corrects the data manipulations that subsequently came to light, but were unknown to (most) observers at that time.[8]

"Best levels" data creates more problems, because the irregular updating of time series creates artificial dynamics in the measured data that is not present in the underlying object. The best example here is "Civilian noninstitutional population, 16 and over" (FRED:

[8] For a famous example of the use of real-time data, see Orphanides (2001).



Figure 4: The top panel depicts the population level in 1000s (FRED: CNP16OV). The bottom panel reports its growth rate in percent. The blue line is the growth rate of the raw series while the red line is the HP-filtered trend with smoothing parameter $\lambda = 10{,}000$.

CNP16OV/BLS: LNS10000000), which is the LNSINDEX from Smets and Wouters (2007) and has been widely used in the subsequent literature. Population itself is a slow-moving, smooth object, which seems to be confirmed by looking at the level of this series in the top panel of Figure 4. However, as pointed out in Edge, Gürkaynak, and Kisacikoğlu (2013), the level data provides a misleading picture. Looking at the growth rate in the bottom panel of the same figure, pronounced quarterly spikes of sometimes more than 1 percent appear. The reason for these spikes is not regular population dynamics, but the "best levels" data construction. Whenever decennial censuses or annual benchmarking of the Current Population Survey (CPS) become available, the current and subsequent data points are updated to reflect this new information. Hence, the spikes in 1990 and 2000 simply reflect the newly arrived census population estimates and not any economically relevant sudden changes in actual population. Using such a data series to compute per capita values can potentially introduce spurious dynamics at business cycle frequency that are an artifact of measurement. Edge, Gürkaynak, and Kisacikoğlu (2013) thus recommend:

1. using a smoothed value of this population series to transform variables like GDP or investment, which do not have the same forward-basis revisions, to per capita values. The red line in the bottom panel of Figure 4 shows the smoothed population series derived from using an HP-filtered trend with smoothing parameter 10,000. I use this

filtered series as a substitute for the unavailable smoothed population series from the Federal Reserve's FRB/US model used in Edge, Gürkaynak, and Kisacikoğlu (2013).[9]

2. using the unsmoothed value of this series to transform variables from the same source that have the same forward-basis revisions. For example, the Smets and Wouters (2007) "Civilian Employment, Persons 16 years of age and older" series (FRED: CE16OV) is also derived from the CPS so that censuses and benchmarkings show up in this series. Simply filtering this employment series would also eliminate the actual economic dynamics at business cycle frequency. As only employment per capita is entered into the model, one can use the unsmoothed population series to transform the unsmoothed employment series to per capita values. Because both series come from the same source and reflect the same base series changes, the spikes introduced by forward-only updates cancel. As can be seen by comparing employment growth (top panel of Figure 5) with employment per capita growth (bottom panel), forming the ratio of CE16OV over CNP16OV eliminates for example the census peak in 2000.

However, it is important to keep in mind that those solutions only works if the "best levels" data to be smoothed does not contain important business cycle frequency (case 1) or when the "best levels" data has the same underlying basis *and* only enters as a ratio so that the update spikes cancel (case 2). If this were not the case, the only alternative would be to use a different series that does not suffer from the same problems.

Remark 7 (Which Observables for Estimation?)
A thorny issue is which observables to use when estimating a model. There are not many generally agreed-upon guidelines and hardly any research on this topic (Guerron-Quintana 2010, being an important exception). Some good principles to follow are

1. Feed in observables that restrict the behavior of the model features you are interested in or that are new to your model. For example, if you are estimating the business cycle contribution of investment-specific technology shocks, you should use the relative price of investment as an observable. Justiniano, Primiceri, and Tambalotti (2011) for example did not use the relative price of investment as an observable and found investment-specific technology shocks to be an important contributor to business cycle fluctuations. However, Schmitt-Grohé and Uribe (2012) pointed out that the volatility for the relative price of investment implied by the model is magnitudes larger than in

[9] Note that here, in contrast to Remark 12, the use of a two-sided filter is advocated, because we actually want future information to be incorporated in the data estimates.



Figure 5: The top panel depicts the Employment Growth rate in percent (FRED:CE16OV). The bottom panel reports the growth rate of Employment per capita (FRED:CE16OV/CNP16OV) in percent.

the data. In contrast, using the relative price of investment as an observable restricts its model-implied behavior and results in assigning a smaller role of investment-specific technology shocks for business cycles.

2. A different side of the same coin is using data that aids identification of the parameters you are interested in. Even if all your parameters in the model are theoretically identified, they may not be identifiable given the observables. Dynare provides diagnostics for this via the `identification`-command (see e.g. Pfeifer 2013; Ratto and Iskrev 2011).[10]

3. Use data that have a good signal-to-noise ratio, i.e. that are precisely measured. This applies for example to to most national account data.

4. Be careful with data that is only poorly measured with a lot of noise and error. A typical example are average wages (see e.g. Justiniano, Primiceri, and Tambalotti 2013). In that case, consider allowing for measurement error in those variables.

[10] See Canova and Sala (2009) and Iskrev (2010, 2011) on the issue of identification in general.

3.2 Output

Economic models typically feature real output as a variable that is to be matched. Figure 2 shows *Real GDP in Billions of Chained 2005 Dollars* (FRED: gdpc96). It is apparent that Y_t^{data} grows over time. This is the reason we cannot directly enter it into the model as it has no well-defined steady state around which to approximate. Typically, there are two sources of this growth: i) growth in the population N_t and ii) technological growth X_t (just think about the Solow model). But most models focus on business cycles and abstract from long-term movements in the labor force as we did when setting up the RBC-CME model and assumed everything to be in per-capita terms.[11] Thus, the empirical counterpart to the model variable Y_t (output per worker) in equation (2) is actually Y_t^{data}/N_t^{data}.

Remark 8 (NIPA Tables: At Annual Rates)
The U.S. NIPA differ from many other countries' national accounts in that the data are reported "at annual rates", i.e. what the corresponding number would have been over a full year. For example, if quarterly GDP was 100 apples, the NIPA tables will report a quarterly GDP "at annual rates" of $4 \times 100 = 400$ apples. When working with percentage deviations from a BGP or growth rates, this is not an issue as the base relative to which the percentages or rates are computed is also "at annual rates". For example, say a recession hits and only 99 apples instead of 100 are produced per quarter, corresponding to 396 instead of 400 at annualized rates. The absolute deviation of GDP from its BGP is 1 apple per quarter or 4 apples at annualized rates. However, in percentage terms, the deviation is $99/100 - 1 = 396/400 - 1 = -1\%$, regardless of whether data at annualized rates is used or not.
However, it is import to keep the presentation of NIPA data "at annualized rates" in mind if you ever find yourself in a position where you want to use actual level data at quarterly frequency - and not just some form of growth rates.

To get this, we divide GDP Y_t^{data} for example by N_t^{data} in the form of the civilian labor force (FRED: CLF16OV).[12] The result is shown in Figure 6. This provides data for real GDP per capita Y_t^{data}/N_t^{data}, which corresponds to Y_t in most economic models. However, Y_t still grows due to the presence of technology growth. In the following, we will denote Y_t^{data}/N_t^{data}

[11] For problems when neglecting low frequency movement, see e.g. Francis and Ramey (2009).

[12] This essentially corresponds to detrending the model with the labor force N_t, just like the detrending with X_t as we do later in Section 5.



Figure 6: The solid line depicts US real GDP per capita Y_t^{data}/N_t^{data}.

with $Y_{pc,t}^{data}$ for per capita and will show how to match it to output y_t in the Dynare model.

Remark 9 (An Additional Complication: Chained Indexes)
The statistical offices of many countries have shifted to the use chain-weighted real data in their national accounts. The great advantage of chain weighting is that it limits the substitution bias inherent in fixed weight estimates. Let's say we have a two-sector economy whose output consists of investment goods (computers) and consumption goods. If the relative price of investment goods decreases (for example due to rapid productivity growth à la Moore's Law), people will typically buy more of them. Hence, using the prices of a fixed base year where investment goods were relatively more expensive tends to overstate output growth (see e.g. Jones 2002; Whelan 2002). However, limiting this substitution bias through chain weighting comes at a cost: the loss of additivity. While the national income accounting identity

$$Y_t^{nom} = C_t^{nom} + I_t^{nom} \tag{18}$$

always holds in nominal terms, the same is not true in real terms if you are using chained indexes (except for the base year). That is

$$Y_t^{real} \neq C_t^{real} + I_t^{real} \,. \tag{19}$$

The reason is that the real components of output like C_t^{real} and I_t^{real} are measured in different units, i.e. consumption and investment goods, respectively, and that their relative price, i.e. the conversion rate between the two types of goods is changing over time.

Rather, what would be needed is something along the lines of

$$Y_t^{real} = C_t^{real} + A_t I_t^{real} \,, \tag{20}$$

where A_t is the relative price of investment goods in terms of the final good and where I am assuming that the consumption good is identical to the final good so that the relative price is 1.

However, most economic models are one-good models and have a resource constraint of the type shown in equation (19). Using chain-weighted data as observables to proxy for the real variables in such a one-good economy would neglect that the data for the sub-aggregates of output are measured in different types of goods. Comparing the stylized equations (19) and (20) shows that such an approach would effectively neglect the relative price A_t. Note also that using growth rates does not help as the growth of A_t would be neglected.
The literature has found three ways to deal with this problem:

1. Take the one-good model structure seriously and impose that all goods are measured in units of the final good. In that case, one starts from nominal national account values and, as implied by the model, uses the price of the final good, i.e. GDP deflator, as the deflator. This is the procedure used in e.g. Smets and Wouters (2007) or Schmitt-Grohé and Uribe (2012). Effectively, one does not use chain weighted data at all - except for computing the GDP deflator.

2. Model a more complicated multi-sector structure (see Whelan 2003, for a discussion why this might be necessary) that accounts for relative price changes. Then map the real model variables measured in terms of the respective goods to the corresponding chain weighted real data that is measured in the same goods (see e.g. Edge, Laubach, and Williams 2007; Iacoviello and Neri 2010; Ireland and Schuh 2008)

3. Take a mixture of the previous two approaches. That is, use nominal variables deflated with the GDP deflator, but account for the relative price of investment by using it as an additional observable (e.g. Born, Peter, and Pfeifer 2013; Schmitt-Grohé and Uribe 2012)



Figure 7: The solid line depicts the Gross Inflation Rate, Π_t^{obs}, computed as the ratio of the Consumer Price Index (CPI) in two subsequent quarters.

3.3 Inflation

In contrast to output, variables like gross inflation Π_t and the nominal interest rate R_t have direct nontrending/stationary equivalents in the data. Gross inflation Π_t^{data} is usually computed as the ratio of the consumer price index P_t^{data} in two subsequent time periods (FRED: CPIAUCSL):

$$\Pi_t^{obs} = \frac{P_t^{data}}{P_{t-1}^{data}} \tag{21}$$

If the price index used is the quarterly CPI, Π_t^{obs} will be a quarterly gross inflation rate. This is exactly the frequency required by a model in quarterly frequency.[13] The resulting series, Π_t^{obs} is shown in Figure 7.

For some models, it is useful to work with the net inflation rate instead as an observable. The net inflation rate is usually defined as the gross inflation rate minus 1 or alternatively by taking the logarithm of gross inflation rate,

$$\pi_t = log(\Pi_t) \ , \tag{22}$$

which is approximately the same as subtracting 1 if gross inflation is close to 1. Hence, it is typically computed as the log difference of the consumer price index P_t^{data} in two subsequent

[13]This is different from the way inflation is colloquially quoted, which is in annual terms. People remember inflation at the beginning of the 1980s being almost 16 percent per year, not 4 percent per quarter.



Figure 8: The solid line depicts the log-difference of the CPI, i.e. the logarithm of the gross inflation rate. It is quoted as quarterly percentage change.

time periods:

$$\pi_t^{obs} = \log\left(\frac{P_t^{data}}{P_{t-1}^{data}}\right) = \log P_t^{data} - \log P_{t-1}^{data} \tag{23}$$

Alternatively, one could use true growth rates

$$\pi_t^{obs} = \frac{P_t^{data} - P_{t-1}^{data}}{P_{t-1}^{data}} \tag{24}$$

For small inflation rates, both formulations are approximately equivalent. The resulting series is displayed in Figure 8.

3.4 Nominal Interest Rate

Observed interest rates are also stationary,[14] but are often quoted as i) *net* interest rates in percentage points and ii) in annualized form. In contrast, most business cycle models are written in quarterly frequency and consider *gross* interest rates. Figure 9 shows the effective Federal Funds rate (FRED: FF). This annualized net interest rate R_t^{data}, quoted in annualized percentage points, has to be transformed into a quarterly gross interest rate R_t^{obs} to conform with our model shown in Listing 1. This is typically done by dividing R_t^{data} by four hundred

[14] This holds clearly true for real interest rates. For nominal interest rates the evidence is not as clear-cut, but at least for most developed economies where the (Generalized) Taylor principle should have been satisfied, they should also be stationary (Clarida, Galí, and Gertler 2000; Davig and Leeper 2007).



Figure 9: The solid line depicts the effective Federal Funds Rate. It is quoted as a net annual interest rate in percentage points.

and adding 1:

$$R_t^{obs} = 1 + \frac{R_t^{data}}{4 \times 100} \tag{25}$$

Again, this is an approximation of the correct geometric mean:

$$R_t^{obs} = \left(1 + \frac{R_t^{data}}{100}\right)^{\frac{1}{4}} \tag{26}$$

Equation (25) will transform a net annualized rate of 4% into a quarterly gross interest rate of 1.01. The resulting time series of R_t^{obs} is shown in Figure 10.

Remark 10 (Scaling With a Factor 100)

In case of a log-linear(ized) model, it is not uncommon to multiply all observables, y_t^{obs}, by a factor 100. Consider the linear version of the state-space representation in equations (16)-(17):

$$x_t = g_x x_{t-1} + g_u \varepsilon_t^{struct} \tag{27}$$

$$y_t^{obs} = h_x x_t + h_u \varepsilon_t^{obs} \; . \tag{28}$$



Figure 10: The solid line depicts the effective Federal Funds Rate as a gross quarterly interest rate.

Multiplying both equations by 100 implies

$$100 \times x_t = g_x \times 100 \times x_{t-1} + g_u \times 100 \times \varepsilon_t^{struct} \tag{29}$$

$$100 \times y_t^{obs} = h_x \times 100 \times x_t + h_u \times 100 \times \varepsilon_t^{obs} \; . \tag{30}$$

Thus, due to linearity, this multiplication will scale everything including the shock processes by 100 and allows redefining $\check{x}_t \equiv 100x_t, \check{y}_t^{obs} \equiv 100y_t^{obs}, \check{\varepsilon}_t^{struct} \equiv 100\varepsilon_t^{struct}$ and $\check{\varepsilon}_t^{obs} \equiv 100\varepsilon_t^{obs}$. The resulting new state space representation is linear in the redefined variables and differs from the first one in that it is directly interpretable as percentages. For example, while for the untransformed variable a 1% deviation from steady state was equal to $\hat{x}_t = 0.01$, for the retransformed variable 1% will be equal to $\check{\hat{x}}_t = 1$. It is important to keep in mind that this scaling will also affect the shock variances and impulse responses. A shock standard deviation for $\check{\varepsilon}_t^{struct}$ of 0.01 will imply a standard deviation of 0.01% (and not of 1% as was the case for ε_t^{struct}). Most importantly, this implies that the prior distributions over shock standard deviations have to be adjusted accordingly if this transformation is used. Otherwise, if you forget to adjust them, the data will seem to be 100 times as volatile as implied by your prior.
Three warnings are in order.

1. Be careful when only multiplying some observables by a factor of 100. Typically, this will lead to wrong results as the differential size of the observables is inconsistent with

the cross-equation restrictions of the model. You might end up with implications like a $1 = 100\%$ percent shock to TFP suddenly implying only a $0.01 = 0.01\%$ change in interest rates.

2. Keep in mind that this transformation is only valid if you are dealing with a linear type of model. If you are using a higher order approximation instead, you will not be able to use this type of variable transformation as e.g. $100 \times x_t^2 \neq (100 \times x_t)^2$. You will have to work with the untransformed variables.

3. Unconsciously scaling by 100 is a common source of mistakes. For example, if you have a shock `eps_z` in your model that denotes the percentage deviations of a variable from its steady state and you use

```
shocks;
var eps_z; stderr 1;
end;
```

you are implicitly using a shock size of 100%. As noted above, this is fine if your model is linear. But if you are using higher order perturbation techniques, this large shock size might lead to problems, including explosive simulation paths if you do not use pruning (see Kim et al. 2008).

4 Models Without a Specified Trend

After those rather general data transformations, we are now in a position to start mapping the data to actual Dynare model variables. For this purpose, we need to distinguish according to the type of trend assumed to underlie the model. In Section 5 we will look at models that explicitly specify the trend process, but first we start with considering models that do not have a specific trend, i.e. that completely abstract from movements in the BGP due to labor augmenting technology growth X_t. Instead, we assume a stationary model is written down that only describes the behavior of the economy around the BGP/growing steady state. Given that the model variables are assumed to represent stationary concepts, the detrending of the empirical counterparts to those model variables has to take place outside of the model. How this can be done will be detailed in this section.

In order to have an example to work with, we consider the introductory model with its processes $X_t = 1 \, \forall \, t$ and its assumption that TFP follows

$$A_t = e^{z_t} \tag{14}$$
$$z_t = \rho z_{t-1} + \varepsilon_t, \; \varepsilon_t \sim \mathcal{N}(0, \sigma^2) \tag{15}$$

In this case, A_t can be interpreted as the fluctuation of technology around its (unspecified) implicit long-term trend. A_t is a stationary log-normal process, implying all our model variables are stationary as well. Consistent with our notational convention to denote the stationary equivalent to trending variables with lowercase letters, the FOCs can be written as[15]

$$\frac{1}{c_t} = \beta E_t \frac{1}{c_{t+1}} \left[\alpha e^{z_{t+1}} k_t^{\alpha-1} + (1-\delta)\right] \tag{31}$$
$$\frac{1}{c_t} = \beta E_t \frac{1}{c_{t+1}} \frac{R_t}{\Pi_{t+1}} \tag{32}$$
$$c_t + k_t - (1-\delta)k_{t-1} + \frac{b_t}{P_t} = e^{z_t} k_{t-1}^{\alpha} + \frac{b_{t-1}}{P_{t-1}} R_{t-1} \frac{1}{\Pi_t} \tag{33}$$
$$z_t = \rho z_{t-1} + \varepsilon_t \tag{34}$$

These equations are the source of the mod-file presented in Listing 1 and the reason we used lowercase letters.

As the model is agnostic about the source of the trend in the data, the researcher has relatively large leeway about the econometric detrending of the data. We will first describe ways to detrend the data, before describing how to specify the mapping between the detrended data and the model variables.

4.1 Detrending Data

Common ways of getting the trend out of trending variables like output are:[16]

1. One-sided HP-filter (Stock and Watson 1999)
2. Linear-(quadratic) trend (see e.g. King and Rebelo 1999)
3. First-difference filter (see e.g. Smets and Wouters 2007)

[15] Note that nominal bonds B_t contain two sources of non-stationarity. First, they inherit the growth trend of the economy deriving from labor augmenting technology. After detrending by this source of non-stationarity, we get nominal detrended bonds b_t. Second, however, these nominal bonds still inherit the non-stationarity of the price level P_t. This is the reason we still have to detrend b_t by the price level.

[16] There are of course many others. Here, I only list the most common ones.

(4. Hodrick Prescott filter (Hodrick and Prescott 1980), see Remark 12)[17]

(5. Baxter-King (bandpass) filter (Baxter and King 1999), see Remark 12)

Typically, it is not the series $Y_t^{pc,data}$ that is filtered, but the logarithm $\log(Y_t^{pc,data})$. The reason is that taking logs makes the resulting series scale invariant, which is important with exponentially growing variables.

Remark 11 (Log-levels: Example)
Say steady state output grows along the BGP from $\bar{Y}_0 = \$1,000$ to $\bar{Y}_1 = \$10,000$. Say in both cases an oil price shock leads to a recession with actual GDP being 1% below steady state, i.e. $Y_0 = \$990$ and $Y_1 = \$9,900$. When considering absolute deviations, the same recession size of 1% leads to a deviation of output from its trend of \$10 in $t = 0$ and \$100 in $t = 1$. Hence, when looking at absolute deviations measured business cycles would be growing over time as trend growth is not properly taken into account. In contrast, when using logarithms: $log(\bar{Y}_0) = 6.91$, $log(\bar{Y}_1) = 9.21$, $log(Y_0) = \$6.90$, and $Y_1 = \$9.20$. Thus, the difference in logs is 0.01 in both cases.

Remark 12 (Non-Causal Filters)
Thou shalt not use a non-causal, i.e. two-sided filter like the HP-filter (Hodrick and Prescott 1980) or the Baxter-King filter (Baxter and King 1999) when detrending data for DSGE estimation (or Vector Autoregression (VAR) estimation for that matter). Your model solution will take the form of a backward looking state-space system, $x_t = g(x_{t-1}) + \varepsilon_t^{struct}$ i.e. the solution today depends only on current and past states and shocks. However, the two-sided HP-filter is a non-causal filter that takes values from y_{t-3}^{unfilt} to y_{t+3}^{unfilt} to construct y_t^{filt}. Like a simple moving average it takes future values to construct current filtered data. This contradicts the backward looking structure of the model solution. The better option is to use the backward-looking one-sided HP-filter (Stock and Watson 1999).

Unfortunately, different ways of detrending lead to different characteristics of the implied economic cycles (e.g. linear detrending implies more persistent deviations from trend than one-sided HP-filtered data unless the smoothing parameter is infinity). The actual choice of

[17]The Hodrick-Prescott (HP)-filter is an example of "Stigler's law of eponymy" stating that no scientific discovery is named after its original discoverer. It is thought to have originally been developed by Whittaker (1922).



Figure 11: The solid line depicts growth rates of US real GDP per capita (first-difference filter). The red dashed line shows the unconditional mean, which is different from 0.

the type of detrending is usually guided by the prior of the researcher on which technique best filters out trends unrelated to business cycles while at the same time preserving the data characteristics actually related to business cycles.[18]

From the perspective of matching detrended data to the model, the different detrending techniques differ mostly with respect to one important characteristic: whether they also demean the data or not. For example, data filtered with the one-sided HP-filter will always have (approximately) mean 0 (see Figure 12),[19] while data in first differences will still have the average growth rate over the sample as its mean (see Figure 11). This has important implications on the observation equation to specify.

Remark 13 (Forgetting about the Mean: A Warning)
If the model variable y_t to which the observable y_t^{obs} is matched has mean 0, for example because it is a deviation from steady state, and you match it to a variable that does not have mean 0 without accounting for this, you will erroneously force the shocks to account for a positive mean in the observed series.

[18] For a glimpse into the big discussion on the effects of filtering, see the introduction to Canova and Ferroni (2011) and articles of Canova (1998a,b) and Burnside (1998).

[19] The Kalman filter used for one-sided HP-filtering includes a zero constant so that the filtered series will be asymptotically mean 0. For small samples, a small mean may still be present.



Figure 12: The solid line depicts log deviations of GDP from its one-sided HP-filtered trend. The red dashed line shows the unconditional mean.

Remark 14 (Cointegration Relationships)
Sometimes the model builder wants to impose the presence of a certain cointegration relationship. For example, the one-sector neoclassical growth model implies detrending all variables with the same long-term trend instead of detrending each variable with its own in-sample trend. The idea is that if the econometrically determined "in-sample trend" in e.g. consumption differs from the one of output, this must be explained by shocks that differently affect output and consumption, because theory otherwise predicts that the underlying trends are the same. The most common form to impose such a relationship is using growth rates, i.e. a first-difference filter, and then using the mean output growth rate to also demean the growth rate of consumption, investment, etc. (see e.g. Schmitt-Grohé and Uribe 2012)

4.2 Log-linearized Models

A typical application are log-linearized DSGE models, where all variables are in percentage deviations from their deterministic steady state (i.e. you can use the `model(linear)`-option of Dynare). In this case,

$$\hat{y}_t \equiv \log(y_t) - \log(\bar{y}) \ , \tag{35}$$

where y_t denotes a generic variable in intensive form, hats denote variables in percentage deviations and bars indicate steady state values. In case of a log-linearized model, the steady state values of the original y_t are usually provided and used as parameters, while the $\hat{y}_t$ are the actual variables used and entered into Dynare. In this case, the steady state of all variables $\hat{y}_t$ is 0. A log-linearized version of our model in Listing 1 is shown in Listing 3. As can be seen, the actual steady state values of the economic model variables are used as parameters, while the steady state of the model variables, provided in the `steady_state_model`-block, is 0 (the percentage deviation of a variable from steady state is obviously 0 in steady state). The goal, of course, is to link observed variables to model variables. This is usually straightforward as there is a natural equivalence between the model concept of deviations from steady state and cyclical fluctuation in the data around a trend.

Listing 3: Log-linearized baseline model

```
var y_hat c_hat k_hat A_hat R_hat Pi_hat;
varexo eps_A;

parameters beta delta alpha rhoA phi_pi Rbar Pibar;

alpha   = 0.33;     // capital share
delta   = 0.025;    //deprecation rate
beta    = 0.99;     //discount factor
Pibar   = 1;
Rbar    = 1/beta;
phi_pi  = 1.5;
rhoA    = 0.97;     //TFP autocorr. from linearly detrended Solow residual


model(linear);
#k_ss=((1/beta-(1-delta))/alpha)^(1/(alpha-1));
#y_ss=k_ss^alpha;
#c_ss=y_ss-delta*k_ss;
-1/c_ss*c_hat=-1/c_ss*c_hat(+1)
            +beta*1/c_ss*alpha*k_ss^(alpha-1)*(A_hat(+1)+(alpha-1)*k_hat);
-c_hat=-c_hat(+1)+R_hat-Pi_hat(+1);
y_ss*y_hat=c_ss*c_hat+k_ss*k_hat-(1-delta)*k_ss*k_hat(-1);
y_hat=A_hat+alpha*k_hat(-1);
R_hat=phi_pi*Pi_hat;
A_hat=rhoA*A_hat(-1)+eps_A;
end;

steady_state_model;
k_hat=0;
y_hat=0;
c_hat=0;
```

```
R_hat=0;
Pi_hat=0;
A_hat=0;
end;

shocks;
var eps_A=0.0068^2;
end;

steady;
check;

stoch_simul(order=1,irf=20,periods=250);
```

Remark 15 (initval vs. steady_state_model vs. steadystate-file)
If you do not provide Dynare with any information on steady state values, by default it tries whether 0 is the steady state, and if not, starts a numerical solver to find the steady state. Of course, if you are using a non-linear model and don't provide explicit information about the steady state, you will most probably run into trouble: all variables including consumption will be initialized to 0. But zero consumption is typically incompatible with the functional form assumed for utility, resulting in NaN (e.g. $\log(0) = NaN$). Thus, *unless you are using a linear model, always provide explicit steady state information.* There are three ways to do this:

1. The ideal case is that you computed the steady state analytically using pencil and paper. In this case, you can provide this analytic steady state to Dynare using the `steady_state_model`-block. Dynare will try the values inside of the block and will throw out an error if those values are not consistent with a steady state of the entered model equations. If this happens, you either made a mistake in computing the steady state or in entering the model equations.

2. If you don't know the steady state, you should provide initial guesses to start the numerical optimizer using the `initval`-block. Those values do not need to solve the steady state, but should be as close as possible so as to speed up the numerical routine and to increase the chances of finding the steady state. Even if you have no clue what the steady state is, economic intuition often already provides good guidance. For example, many economic variables cannot be 0 and are positive. Moreover, $I < C < Y < K$, with $I = \delta K$. Labor is often in the range of 1/3. Considering those simple relationships

often helps to improve the initial guess.

3. You can use an explicit steady state file, which is an external Matlab-file that must conform with a certain structure and naming convention (see e.g. the `NK_baseline.mod` in the Dynare examples folder). In this steady state file, you must provide the exact steady state values as in the case of the `steady_state_model`-block. While this seems to be a lot more work than the latter, the advantage of an explicit steady state file is its flexibility. First, it enables users to essentially call any other Matlab function. In case of `NK_baseline.mod`, the steady state has been simplified up to a non-linear equation in labor. The steady state file now allows to call a numerical solver to solve this equation and then provide the analytical steady state values. Second, the steady state file allows for changing parameters to take parameter dependencies into account without resorting to model-local variables. For example, users often want to set labor in steady state to 1/3, but this requires setting labor disutility parameter to a corresponding value. This parameter might in turn depend on the discount factor `beta` as encountered with `Rbar` in Listing 1. Inside of the steady state file, all parameter values can be accessed and reset to the need of the user. But remember: with great power comes great responsibility. The additional flexibility offered by a steady state file increases the scope for errors.

Remark 16 (Nested model-local variables and nested parameters)
Listing 3 correctly defines the steady state values as model-local variables depending on the deep parameters of the model. As detailed in Remark 4, this assures that those values are correctly updated. You might have noticed that the model-local variables are nested, i.e. the model-local variables `y_ss` depends on the model-local variables `k_ss`. Such nesting is possible as long as it is recursive as opposed to circular: while `y_ss` depends on `k_ss`, the latter only depends on deep parameters and is evaluated before `y_ss`. That is, before `y_ss` is encountered first, `k_ss` is already known. Due to this, `y_ss` is already known before `c_ss` is encountered first. It is *not* possible to specify a simultaneous equation system in model-local variables and let Dynare solve them. E.g. you could not define `k_ss` to depend on `y_ss` and then define `y_ss` to depend on `k_ss`. You would have to solve such an equation system yourself.
The same applies to parameters. You cannot access a parameter before it has been defined to define another parameter. This will result in the parameter being `NaN` and Dynare providing

a warning that the parameter has not been correctly initialized.

4.2.1 Output

Consider the case of output. Our model variable $\hat{y}_t$ represents log output deviations from the long-term trend and has mean 0. Thus, it exactly corresponds to the logarithm of empirical output per capita, detrended using any of the above filters that also takes out the mean. We denote this detrended log output variables with y_t^{obs}.[20] Think of y_t^{obs} for example as one-sided HP-filtered log GDP per capita. In this case, specifying the observation equation is basically redundant as we directly "observe" $\hat{y}_t$:

$$y_t^{obs} = \hat{y}_t \tag{36}$$

A Dynare template for our log-linear model using detrended mean zero data is shown in Listing 4.[21]

Listing 4: Observation Equations for Log-Linearized Model using Detrended mean zero data

```
var y_hat R_hat Pi_hat... Pi_obs R_obs y_obs;
model;
...
Pi_obs=Pi_hat;
R_obs=R_hat;
y_obs=y_hat;
end;

varobs Pi_obs R_obs y_obs;
```

4.2.2 Inflation

Next, we consider gross inflation Π_t. If the model is log-linearized, $\hat{\Pi}_t$ used in the model will be the percentage deviation of gross inflation from its steady state (which is approximately net inflation). Figure 13 shows the model inflation series from a log-linearized model. The data corresponding to $\hat{\Pi}_t$ will be:

$$\Pi_t^{obs} = \log\left(\Pi_t^{data}\right) - \log\left(\bar{\Pi}\right) = \hat{\Pi}_t \tag{37}$$

[20] The case of first differences, which does not demean the data will be considered later in Section 5.

[21] As $y_t^{obs} = \hat{y}_t$ you could also drop the additional equation and just specify $\hat{y}_t$ as observed.



Figure 13: The solid line depicts the Inflation Rate *Pi_hat* simulated from our log-linearized model. It is quoted as the deviation of the gross quarterly inflation rate from its steady state.

The tricky issue for the data computations here is to compute the percentage deviations from the steady state $\bar{\Pi}$ in the data as the latter is usually unknown. There are two common ways to deal with this issue.

First, you could use the same detrending procedure used for output, with the resulting series being interpreted as percentage deviation of gross inflation from a time-varying steady state/trend, $\Pi^{data}_{trend,t}$ (i.e. implicitly using $\bar{\Pi} = \Pi^{obs}_{trend,t}$):

$$\Pi_t^{obs} = \log\left(\Pi_t^{data}\right) - \log\left(\Pi^{data}_{trend,t}\right) = \hat{\Pi}_t \tag{38}$$

The resulting series is shown in Figure 14.

Second, as in Figure 15 you could assume that the steady state in the data corresponds to the long-run mean (i.e. implicitly using $\bar{\Pi} = \text{mean}\left(\Pi_t^{obs}\right)$)[22], meaning that we use[23]

$$\Pi_t^{obs} = \log\left(\Pi_t^{data}\right) - \text{mean}\left(\log\left(\Pi_t^{data}\right)\right) = \hat{\Pi}_t \tag{39}$$

[22] A tricky issue here is Jensen's Inequality. Using log-differences is only up to first order equivalent to using percentage deviations. As a consequence, if we were using $\bar{x} = \text{mean}\left(x_t^{obs}\right)$ to compute $\hat{x}_t = \log x - \log \bar{x}$, the resulting series for $\hat{x}_t$ would not be mean 0. Therefore, we ignore Jensen's Inequality and interchange the log and the mean-operator to use $\hat{x} = \log x_t^{obs} - \text{mean}(\log x_t^{obs})$. This results in a mean 0 series.

[23] This version is equivalent to using the demeaned net inflation rate.



Figure 14: The solid line depicts the log deviation of gross inflation from its one-sided HP-filtered trend.

Which version you use only affects the data treatment, but not how the observation equation needs to be specified. In both cases the Π_t^{obs} in equations (38) and (39) corresponds to $\hat{\Pi}_t$, implying an observation equation of the form specified in Listing 4.

4.2.3 Interest Rate

For the nominal interest rate R_t, the treatment is similar to the one for inflation Π_t. If the model is log-linearized, $\hat{R}_t$ used in the model will be the percentage deviation of the quarterly gross interest rate from its steady state (which is approximately the net quarterly interest rate[24]). The data corresponding to $\hat{R}_t$ will be:

$$R_t^{obs} = \log\left(1 + \frac{R_t^{data}}{4 \times 100}\right) - \overline{\log\left(1 + \frac{R_t^{data}}{4 \times 100}\right)} = \hat{R}_t \tag{40}$$

The tricky issue for the data computations again is to compute the percentage deviations from the steady state $\bar{R}$ in the data as the latter is usually unknown. The two common ways to deal with this issue from the previous subsection also apply here.

[24]While this is a convenient way to think about it, it is not a practical way to specify the observation equation. The percentage deviation of the gross interest rate from steady state in steady state will be 0, but the net interest rate will be not. This small difference in mean will have to be accounted for by shocks. Thus, construct the observed data series by proper demeaning!



Figure 15: The solid line depicts the log deviation of gross inflation from its sample mean.

First, you could use the same detrending procedure used for output, with the resulting series being interpreted as percentage deviation of the gross interest rate from its time-varying steady state/trend, $R^{data}_{trend,t}$ (i.e. implicitly using $\bar{R} = R^{data}_{trend,t}$):

$$R^{obs}_t = \log\left(1 + \frac{R^{data}_t}{4 \times 100}\right) - \log\left(R^{data}_{trend,t}\right) = \hat{R}_t \tag{41}$$

Second, you could assume that the steady state in the data corresponds to the long-run mean (i.e. implicitly using $\bar{R} = \text{mean}\left(\log\left(1 + \frac{R^{data}_t}{4\times 100}\right)\right)$), meaning that we use

$$R^{obs}_t = \log\left(1 + \frac{R^{data}_t}{4 \times 100}\right) - \text{mean}\left(\log\left(1 + \frac{R^{data}_t}{4 \times 100}\right)\right) = \hat{R}_t \tag{42}$$

Which version you use only affects the data treatment, but not how the observation equation needs to be specified. In both cases the R^{obs}_t in equations (41) and (42) corresponds to $\hat{R}_t$, implying an observation equation of the form specified in Listing 4.

4.3 Estimating Parameters Depending on Sample Means in Log-Linear Models

In Section 4.2, we eliminated the data mean. But there are cases where you actually need the data mean of stationary variables. For example, you might want to estimate the discount factor β (see e.g. Fernández-Villaverde 2009) or the steady state inflation rate $\bar{\Pi}$ (see e.g.

Smets and Wouters 2007). In this case, using demeaned data for interest rates or inflation as was done in the previous section is not an option as this would imply the total loss of information about the mean. But there is a simple way to get this information back into the log-linearized model. A simple transformation of equation (37) implies

$$\Pi_t^{obs} = \log\left(\Pi_t^{data}\right) = \hat{\Pi}_t + \log\left(\bar{\Pi}\right) , \tag{43}$$

where Π_t^{obs} is now the non-demeanded net inflation rate and $\bar{\Pi}$ is the steady state gross inflation rate to be estimated. Similarly, for the interest rate R_t, we have:

$$R_t^{obs} = \left(\log\left(1 + \frac{R_t^{data}}{4 \times 100}\right)\right) = \hat{R}_t + \log\left(\bar{R}\right) = \hat{R}_t + \log\left(1/\beta\right) \tag{44}$$

where R_t^{obs} is now the undetrended logged quarterly net interest rate and β is the discount factor to be estimated. The corresponding Dynare code is shown in Listing 5

Listing 5: Observation equations for log-linearized model using data mean to estimate parameter

```
var ... Pi_obs R_obs;
parameter Pibar;
model;
...
Pi_obs=Pi_hat+log(Pibar);
R_obs=R_hat+log(1/beta);
end;

estimated_params;
Pibar ,...;
beta ,...;
end;

varobs ... Pi_obs;
```

4.4 Linearization vs. Log-linearization

After considering log-linear models, we now consider a model where the nonlinear equations are entered into Dynare and Dynare's preprocessor is used for (log)linearization purposes. The first thing to note is that Dynare only performs linearizations, not log-linearizations (the notable exception is detailed in Remark 17). This means that if we want Dynare to perform a log-linearization, we have to make use of the fact that by rewriting $f(x) = f\left(e^{\log x}\right)$ and using a linear Taylor approximation in $\log x$ instead of x about $\log x_0$, we actually get everything in percentage deviations from steady state:

$$
\begin{aligned}
f(x) = f\left(e^{\log x}\right) &\approx f\left(e^{\log x_0}\right) + \frac{\partial f\left(e^{\log x}\right)}{\partial x} \frac{\partial e^{\log x}}{\partial \log x}\bigg|_{x=x_0} (\log x - \log x_0) \\
&= f(x_0) + f'(x_0)\, x_0 \hat{x} \qquad (45)
\end{aligned}
$$

That implies: redefining variables to be in logs by putting them into *exp*() and then doing a standard linearization (thus called log-linearization) results in expressing everything in percentage deviations. Consider the budget constraint of the introductory example in intensive form as we are dealing with a non-linear model while still abstracting from trend growth:

$$
c_t + k_t - (1-\delta)k_{t-1} + \frac{b_t}{P_t} = e^{z_t} k_{t-1}^{\alpha} + \frac{b_{t-1}}{P_{t-1}} R_{t-1} \frac{1}{\Pi_t} \qquad (33)
$$

The way we entered it into Dynare in Listing 1, we are performing a linearization as Dynare is performing a Taylor approximation in the defined variables, which are the levels of consumption, output, etc., not their log-levels.

Remark 17 (Dynare's `loglinear` option)
While Dynare by default only computes linearizations, it also has the option `loglinear` in order to perform log-linearizations at `order=1`. When using this option, all displayed results will be for the logged variables, not the original ones. In the context of estimation, `fs2000.mod` in the `examples` folder of Dynare provides an example.
When using the `loglinear`-option, Dynare will use the Jacobian transformation $1/\bar{x}$ to perform an automatic variable substitution for *all* variables - without exceptions. It is important to keep this in mind as this transformation is only valid for variables with strictly positive steady state. This has two important implications. First, you cannot simply write down your model already including mean zero log-linear processes as this would imply a division by zero in the transformation. Hence, enter your exogenous processes in the form

```
y=z*k(-1)^alpha;
log(z)=rhoz*log(z(-1))+eps_z;
```

instead of

```
y=exp(z)*k(-1)^alpha;
z=rhoz*z(-1)+eps_z;
```

While both formulations are basically equivalent, in the former, `z` represents the level of TFP.

It has steady state 1 so that the `loglinear` option will work. In the latter listing, a log transformation using `exp(z)` has already been applied and `z` is log-TFP with mean 0. In this case, the `loglinear` option with its Jacobian transformation would imply a division by 0. Second, you cannot use the `loglinear` option when you have variables with negative steady state values as is sometimes the case for debt. In this case the Jacobian transformation would be invalid. In both cases, Dynare will provide an error message.
An important principle to keep in mind when using the `loglinear`-option is that Dynare will use the logarithms in the background while you provide information on the original model with the unlogged variables. This applies to both the data and the model variables. For example, you have to provide (initial) steady state values for the original unlogged variables. Similarly, when writing down an observation equation it will be in terms of variables that have not (yet) been logged. Thus, the corresponding data also must not (!) be logged. Rather, Dynare will take the logarithm of both the data and the model variables for you - again treating all variables in the model including the ones on the left-hand side and the right-hand of the observation equations as unlogged variables. You can override the logging of the data by specifying the `logdata`-option in the `estimation`-command.

But we want a log-linearization for the reasons specified in Remark 11 and need to express all variables in logs. We do this by writing for example $y_t = e^{\log(y_t)} = e^{\tilde{y}_t}$ and then using the variable

$$\tilde{y}_t \equiv \log(y_t) \tag{46}$$

in Dynare so that $\tilde{y}_t$ actually represents the logarithm of output, i.e. the log-level. That way, when Dynare does a linear approximation around $\tilde{y}_t$ it actually performs a log-linearization around $\log(y_t)$.[25]

The corresponding Dynare code for a log-linearization is shown in Listing 6.

Listing 6: Nonlinear Model for Log-Linearization

```
var y_tilde c_tilde k_tilde z R_tilde Pi_tilde;
varexo eps_z;

parameters beta delta alpha rhoz phi_pi Pibar;

alpha   = 0.33;     // capital share
delta   = 0.025;    //deprecation rate
```

[25]Usually people don't go through the hassle of explicitly defining a new variable (here the one with a tilde) for the logarithm of the respective variable, but rather put the original y_t in $exp()$ as $exp(y_t)$ and keep the variable name y_t, although, after being put into $exp()$, the new y_t actually represents the old $\log(y_t)$.

```
beta     = 0.99;     //discount factor
Pibar    = 1;
phi_pi   = 1.5;
rhoz     = 0.97;     //TFP autocorr. from linearly detrended Solow residual

model;
#Rbar     = 1/beta;
1/exp(c_tilde)=beta*1/exp(c_tilde(+1))*(alpha*exp(z(+1))*exp(k_tilde)^(alpha
    -1)+(1-delta));
1/exp(c_tilde)=beta*1/exp(c_tilde(+1))*(exp(R_tilde)/exp(Pi_tilde(+1)));
exp(z)*exp(k_tilde(-1))^alpha=exp(c_tilde)+exp(k_tilde)-(1-delta)*exp(k_tilde
    (-1));
exp(y_tilde)=exp(z)*exp(k_tilde(-1))^alpha;
exp(R_tilde)/Rbar=(exp(Pi_tilde)/Pibar)^phi_pi;
z=rhoz*z(-1)+eps_z;
end;

steady_state_model;
k_tilde=log(((1/beta-(1-delta))/alpha)^(1/(alpha-1)));
y_tilde=log(exp(k_tilde)^alpha);
c_tilde=log(exp(y_tilde)-delta*exp(k_tilde));
R_tilde=log(1/beta);
Pi_tilde=log(Pibar);
z=0;
end;


shocks;
var eps_z=0.0068^2;
end;

steady;
check;

stoch_simul(order=1,irf=20,periods=2000);
verbatim;
```

Remark 18 (Log-linearization and the Transformation of Steady States)
The variable redefinition of $\tilde{y}_t = \log(y_t)$ implies that $\tilde{y}$ is now the logarithm of the original model variable y_t. Thus, when you provide initial values or analytical steady state values, keep this transformation in mind. For example, if output y_t in the non-linear model has

steady state 1, the new $\tilde{y}_t$ will have steady state $\bar{\tilde{y}} = \log(\bar{y}) = \log(1) = 0$. This can be clearly seen in the `steady_state_model`-block of Listing 6.

Remark 19 (Variables with Negative Steady States or already in Percent)
There are cases when a log-linearization is not desired or simply infeasible. For example, if you consider government bonds B_t in a model, they might have a negative steady state, meaning that taking the log is not allowed (unless you want to deal with complex steady states). For those variables, one wants a linearization, not a log-linearization. It is straightforward to achieve this in Dynare. If a variable is in levels, Dynare will perform a linearization. If a variable is in log-levels, Dynare will perform a log-linearization. Thus, simply do not put B_t into $exp()$ to keep it in levels.

The same is also possible for variables with positive steady states where you prefer a linearization instead of a log-linearization due to a better interpretability. The reason for using logarithms of all variables usually is that the resulting IRFs for a variable $\tilde{y}$ in response to a shock to a variable $\tilde{x}$

$$\frac{\partial \tilde{y}}{\partial \tilde{x}} = \frac{\partial \log y}{\partial \log x}$$

have the (approximative) interpretation of an elasticity. You get a percentage change in y for a percentage change in x. But what if one of the variables is already in logs or measured in percent? In this case, one actually wants a semi-elasticity of the form

$$\frac{\partial \tilde{y}}{\partial x} \text{ or } \frac{\partial y}{\partial \tilde{x}}$$

because due to either y or x already being in percent, this semi-elasticity truly has the economic interpretation of an elasticity. Consider the clarifying example of a tax rate τ with mean 33%. It is already measured in percent. A linear/absolute deviation of 1%, $\partial \tau$ of its steady state already has the interpretation of 1 percentage point change from 33% to 34%. In contrast, a log-linear/percentage deviation of 1% from steady state, $\partial \log(\tau)$ implies an increase of 1% $\times$ 0.33=0.33 percentage points from 33% to 33.33%. Thus, the "semi-elasticity" of $\partial \tau$ already has our desired interpretation. For this reason, net rates are also often not put into $exp()$.

4.5 Observation Equations in Case of Nonlinear models

The only difference regarding the specification of observation equations for models entered non-linearly in Dynare is to keep in mind that

$$\tilde{y}_t \neq \hat{y}_t \tag{47}$$

That is, the variables entered into Dynare are not log-deviations from steady state, but rather the logarithm of actual model variables (remember $\tilde{y}_t \equiv \log y_t$). Basically they are

$$\tilde{y}_t = \log y_t = \log y_t - \log \bar{y} + \log \bar{y} = \hat{y}_t + \log \bar{y} \tag{48}$$

4.5.1 Output

Equation (48) suggests an easy way to specify an observation equation. We could use a series of percentage deviations from steady state/trend, $\hat{y}_t$, of the type coming from a one-sided HP filter applied to log time series as our observed series y_t^{obs}. This would imply that

$$y_t^{obs} = \hat{y}_t \overset{(48)}{=} \tilde{y}_t - \log(\bar{y}) \,, \tag{49}$$

where $\bar{y}$ is the steady state value of the unlogged original output variable y_t in the model. The right-hand side of equation (49) is just the demeaned log level of the model variable. Note that for a non-linear model, you typically have to provide this value for detrending model variables to Dynare (or estimate it as for $\bar{\Pi}$ in Section 4.3). To do this, you can use the `steady_state()`-operator in the `model`-block. When Dynare for example encounters the code `steady_state(y_tilde)`, it substitutes the computed steady state for $\bar{\tilde{y}} = \log(\bar{y})$. As the steady state is updated whenever the deep parameters are changed, this provides a convenient alternative to using #-model-local variables.

Remark 20 (The `steady_state`-operator in Linear Models)
Using the `steady_state`-operator is not an alternative to the use of model-local variables in linear models to compute the linearization coefficients. The reason is that the `steady_state`-operator allows accessing the steady state values of the endogenous model variables. But those variables are the $\hat{y}_t$ and not the y_t for which we need the steady state values to compute the linearization coefficients.

The observation equation corresponding to the use of detrended and demeaned data with model variables in log-levels is shown in Listing 7. Instead of model-local variables, the steady state operator is used to access the steady state value of $\bar{\tilde{y}} = \log(\bar{y})$.

Listing 7: Observation Equation in Nonlinear Model for Log-Linearization using demeaned data

```
var y_tilde y_obs...;

model;
#R_bar=1/beta;
...
1/exp(c_tilde)=beta*1/exp(c_tilde(+1))*(alpha*exp(z(+1))*exp(k_tilde)^(alpha
    -1)+(1-delta));
y_obs=y_tilde-steady_state(y_tilde);
Pi_obs=Pi_tilde-steady_state(Pi_tilde);
R_obs=R_tilde-steady_state(R_tilde);
end;
```

Using detrended and demeaned data and matching it to the model variables as in Listing 7 avoids that the researcher has to take a stance on matching the levels of model and empirical means. For example, many models are calibrated to a labor share in total time of 1/3. Matching this to the level of actually measured hours worked, which are sometimes quoted as indices (e.g. FRED:HOANBS), is difficult and can be avoided by using such an observation equation where the model mean is subtracted and the empirical variable demeaned.

Remark 21 (Calibration vs. Estimation)
Is is very common to only estimate a subset of parameters and fix the other parameters via calibration. While fixing parameters by calibration seems to contradict the purpose of estimating a model, there are two different cases when calibration is generally considered appropriate:

- A parameter is only weakly identified from the data used, but there is strong external evidence on the value of this parameter (see e.g. Fernández-Villaverde 2009). Typical examples are the risk aversion parameter in the utility function, the capital share used in the production function, and the depreciation rate. The weak identification can also derive from the particular type of data treatment used. When using demeaned data, there is often not enough information left to estimate parameters crucially depending on the mean.

- A parameter is not identified at all and thus needs to be set externally as it cannot be estimated. The most famous case is the substitution elasticity in linearized New Keynesian models, which cannot be identified separately from the Calvo parameter (see Iskrev 2010). In this case, it is customary to fix the substitution elasticity based on extraneous evidence on average firm markups while the Calvo parameter is estimated.

Regarding interpretation, calibration can be seen as a dogmatic point prior. It is equivalent to an estimation without giving the likelihood function the chance to update the mapping from prior to posterior. This has the advantage that the estimation cannot result in the "dilemma of absurd parameter estimates" (An and Schorfheide 2007) for these parameters and is just the most extreme case of using Bayesian priors to prevent this.

The potential disadvantage of fixing parameters by calibration is that the point prior puts all the burden of accounting for the data on the other parameters. Hence, if the fixed parameter is not fixed at the true value, the model will be misspecified and this misspecification will turn up in the other parameter estimates.

Thus, when fixing a parameter based on long-run averages, it is implicitly "estimated" separately from the rest of the parameters by a moment matching (of the mean), because of the (implicit) assumption that this average perfectly identifies the parameter without remaining parameter variability.

4.5.2 Inflation

As with the model that was already entered in log-linearized form, using the observation equation of the type in Listing 7 implies the loss of information about the data mean due to the use of demeaned data. As already mentioned, it might be desired to estimate the steady state level of inflation. Considering that $\tilde{\Pi}_t$ in the non-linear model corresponds to $\log(\Pi_t)$ (as it has no growth trend) and thus is just the logarithm of gross inflation, one can simply use

$$\Pi_t^{obs} = \log\left(\Pi_t^{data}\right) \tag{50}$$

As $\log\left(\Pi_t^{data}\right)$ is actually exactly the definition of log inflation $\tilde{\Pi}_t = \log(\Pi_t)$, the observation equation is

$$\Pi_t^{obs} = \tilde{\Pi}_t \tag{51}$$

The corresponding listing is shown in Listing 8. Two things are noteworthy. First, because the observable and the model variable are identical, one could have simply left out the observation equation and specified `pi` as observed. Second, although `Pi_bar` does not show up in the observation equation, it is contained in the data `pi_obs` and the model variable $\tilde{\Pi}$. The requirement of this mean to be consistent across the data and the model allows for estimating $\bar{\Pi}$.

Listing 8: Observation Equation in Nonlinear Model for Log-Linearization using non-demeaned data

```
var Pi_tilde Pi_obs R_tilde R_obs ...;
parameter Pi_bar beta ...;
model;
#Rbar=1/beta;
...
Pi_obs= Pi_tilde;
R_obs = R_tilde;
end;
```

4.5.3 Nominal Interest Rate

Depending on whether the parameters defining the steady state of the nominal interest rate are estimated or not (e.g. the discount factor or the capital tax rate if present in the model), one can either use an observation equation with demeaned data as in Born, Peter, and Pfeifer (2013) or with undemeaned data (see e.g. Smets and Wouters 2007).

In any case, the procedure is similar to the one for inflation in the previous subsection. As $\tilde{R}_t = \log(R_t)$, we can use the log of the data transformation in equation (25), i.e. use

$$R_t^{obs} = \log\left(1 + \frac{R_t^{data}}{4 \times 100}\right) \tag{52}$$

and simply specify as the observation equation

$$R_t^{obs} = \tilde{R}_t \,. \tag{53}$$

The corresponding observation equation is also shown in Listing 8.

Remark 22 (Filtered vs. Unfiltered Data for Stationary Variables)
In this document, I typically assume that stationary data is entered in non-filtered form, because obviously no filtering is required to render them stationary. However, in principle there is nothing that prevents researches from applying a filter to stationary variables. There may be good reasons for and against doing so:

(+) Sometimes the researcher is interested only in explaining the behavior of the stationary variables in a certain frequency range (say between 8 and 32 quarters). In this case, the treatment of the corresponding observation equation is the same as for non-stationary variables like output: use a filter with a pass-band of 8 to 32 quarters that takes out the other frequencies located in the stop-band of the filter.

(–) By filtering stationary variables you are putting low-frequency components potentially unrelated to growth into the stop-band of your filter. This can sometimes obscure problems in estimation and lead to strange results. More generally, the use of only filtered time series in estimation implies that the frequency components in the stop-band play no role. Sometimes this is desired, but filtering out low-frequency components might for example erroneously allow for a extremely high estimated persistence in the stationary variables, because the effect takes place in the stop-band. For example, labor tax rates are extremely persistent and exhibit near unit-root behavior. In estimation, this might introduce parameter estimates that also imply near-unit root behavior of hours worked. However, because you are filtering out those low-frequency components, those problems for hours might not even pop up in estimation. You will only see that the resulting parameter estimates are implausible.[26] Using at least one unfiltered stationary variable typically gives a more prominent role to low-frequency components.

Remark 23 (Filtering of Stationary Model Variables when Comparing Moments) A priori, it might seem strange to filter the output from a stationary model in order to compare the model moments (variances, covariances) to the data moments. As the model output will be stationary, there seems to be no reason to use a filter in the first place: if for unfiltered variables the second moments will exist. However, one rule that is typically obeyed when comparing data and model moments is to treat both model variables and data consistently in the same way. If you use the HP-filter to extract trend deviations from the data, do so as well in the model (but only use the two-sided HP-filter for comparison, not for estimation; see Remark 12).

The underlying idea for doing so is twofold:

1. Treating both data and the model in the same way makes sure that the moments from both are comparable. Filtered data considers only a particular frequency band that

[26] Note that this problem is typically even more severe if you use indirect inference like simulated method of moments.

is thought of as “business cycle frequencies”. As business cycle models are explicitly built to explain the business cycle, you want it to perform similar in that particular frequency band.

2. By treating the data and the model similarly, you make sure that all distortions and biases that might arise from the data treatment are also present for the model variables when you compare the two. This idea is related to the Sims-Cogley-Nason-approach (see Christiano, Eichenbaum, and Vigfusson 2006).

Remark 24 (Mixing Different Kinds of Filters)
Mixing different types of filters like one-sided HP filtering and first differences in principle poses no problem for estimation. However, you will run into an issue of consistency. When using different filters with different pass-bands, you are essentially assigning different frequency components to the business cycle domain for different variables. Thus, you might have a hard time persuading your readership why you do not use a consistent business cycle definition.

4.6 First-Difference Filter

The previous subsections assumed the use of a filter for non-stationary data that already demeaned the data (and the use of logged unfiltered data for stationary variables). But sometimes people want to use a first-difference filter, which results in non-zero mean data for growing variables, even for models without an explicitly specified trend. I will not cover this case here, because it is relatively rare and a straightforward extension of the use of a first-difference filter in models with an explicitly specified trend as discussed in Section 5. Thus, after consulting that section, the reader should be able to correctly specify the observation equations for this case.

5 Models with Explicitly Specified Trend

While the previous models were agnostic about the source of the trend in the data, there are models where the source of the trend is explicitly specified. Typically, the assumption is the presence of a stochastic growth trend in the form of a random walk with drift. That means the respective variables are integrated, i.e. $I(1)$. Differencing them makes them stationary, i.e.

$I(0)$. This section will thus be mostly concerned with using data in first differences.

5.1 Detrending

A commonly made assumption is that the growth of output per capita results from technological progress that can be represented as labor augmenting technology growth X_t. This technology process X_t is often assumed to follow a random walk with drift

$$X_t = X_{t-1} e^{\Lambda_x + \varepsilon_t^x}, \varepsilon_t^x \sim N\left(0, \sigma_{x,t}^2\right) \tag{54}$$

where Λ_x is the long-run growth rate of labor augmenting technology (and thus of output per capita) and ε_t^x is a permanent shock to technology. Taking logs yields

$$\log X_t = \log X_{t-1} + \Lambda_x + \varepsilon_t^x \tag{55}$$

Consider again our budget/resource constraint, with the law of motion for capital K_t substituted for I_t and the production function for Y_t:

$$C_t + K_t - (1-\delta)K_{t-1} + \frac{B_t}{P_t} = A_t K_{t-1}^{\alpha} X_t^{1-\alpha} + \frac{B_{t-1}}{P_{t-1}} R_{t-1} \frac{1}{\Pi_t} \tag{12}$$

Here, all variables except for R_t and Π_t are growing over time due to X_t growing. A_t still represents a stationary TFP shock.[27] To detrend those variables, denote with lowercase letters non-trending variables and write every generic variable D_t as[28]

$$D_t = d_t X_t \tag{56}$$

Equation (12) can then be written as

$$\begin{aligned} c_t X_t + k_t X_t - (1-\delta) k_{t-1} X_{t-1} + \frac{b_t}{P_t} X_t &= A_t (k_{t-1} X_{t-1})^{\alpha} X_t^{1-\alpha} + \frac{b_{t-1}}{P_{t-1}} X_{t-1} \frac{R_{t-1}}{\Pi_t} \\ c_t + k_t - (1-\delta) \frac{k_{t-1}}{\mu_t} + \frac{b_t}{P_t} &= A_t \left(\frac{k_{t-1}}{\mu_t}\right)^{\alpha} + \frac{b_{t-1}}{P_{t-1}} \frac{1}{\mu_t} \frac{R_{t-1}}{\Pi_t} \end{aligned} \tag{57}$$

[27] There is considerable discussion about the relative business cycle contribution of permanent vs. stationary technology shocks. See e.g. Schmitt-Grohé and Uribe (2012) and Born, Peter, and Pfeifer (2013).

[28] In the presence of non-stationary investment-specific technology growth or other sources of non-stationarity, not all variables have the same trend. In those cases, it is important to detrend every variable with the respective (composite) trend. However, the general procedure still remains the same. For more details on this case, see Born, Peter, and Pfeifer (2013)

where the second line follows from dividing by X_t and where

$$\mu_t = \frac{X_t}{X_{t-1}} \tag{58}$$

is the gross growth rate of technology with

$$\log \mu_t = \log X_t - \log X_{t-1} \overset{(55)}{=} \Lambda_x + \varepsilon_t^x \tag{59}$$

Remark 25 (Linear Trend in Labor Augmenting Technology)
Equation (59) shows that labor augmenting technology growth is an i.i.d.-process without persistence around a mean of Λ_x. As such it nests the case of an exponential trend, i.e. a linear trend in logs for $\varepsilon_t = 0 \,\forall\, t$. By shutting off the shock, technology would grow at the constant rate of Λ_x.

$$\log \mu_t = \log X_t - \log X_{t-1} = \Lambda_x$$

Similarly, consider the Euler equation:

$$\frac{1}{C_t} = \beta E_t \frac{1}{C_{t+1}} \left[\alpha A_{t+1} K_t^{\alpha-1} X_{t+1}^{1-\alpha} + (1-\delta)\right] \tag{6}$$

Using the same detrending trick, we get:

$$\begin{aligned}
\frac{1}{c_t X_t} &= \beta E_t \frac{1}{c_{t+1} X_{t+1}} \left[\alpha A_{t+1} (k_t X_t)^{\alpha-1} X_{t+1}^{1-\alpha} + (1-\delta)\right] \\
\frac{1}{c_t} &= \beta E_t \frac{X_t}{X_{t+1}} \frac{1}{c_{t+1}} \left[\alpha A_{t+1} \left(\frac{k_t}{\frac{X_{t+1}}{X_t}}\right)^{\alpha-1} + (1-\delta)\right] \\
&= E_t \frac{\beta}{\mu_{t+1}} \frac{1}{c_{t+1}} \left[\alpha A_{t+1} \left(\frac{k_t}{\mu_{t+1}}\right)^{\alpha-1} + (1-\delta)\right] \qquad (60)
\end{aligned}$$

Two things are noteworthy. First, all variables in equations (57) and (60) are stationary and have a well-defined steady state as they are in intensive form.[29] Second, equation (59), the law of motion for labor-augmenting technology growth μ_t, is the detrended version of

[29] Of course unless $\alpha = 1$. In this case one would get an AK-model, which, as is well-known, does not imply convergence to a steady state in intensive form.

equation (55) and takes its place as a model equation.

Remark 26 (Different Laws of Motion for Labor Augmenting Technology) Equation (59) can be easily modified to encompass the more general forms of technology growth. For example, to allow for correlated growth rates, one could simply replace it by an Autoregressive (AR)-process e.g.

$$\log \mu_t = (1-\rho_x)\Lambda_x + \rho_x \log \mu_{t-1} + \varepsilon_t^x$$

The corresponding mod-file that includes all the necessary detrending and adjusts the steady state accordingly is presented in Listing 9.

Listing 9: Nonlinear Model with Explicit Trend for Log-Linearization

```
var y_tilde c_tilde k_tilde z R_tilde Pi_tilde mu_tilde;
varexo eps_z eps_x;

parameters beta delta alpha rhoz phi_pi Pibar Lambda_x;

alpha    = 0.33;     // capital share
delta    = 0.025;    //deprecation rate
beta     = 0.99;     //discount factor
Pibar    = 1;
phi_pi   = 1.5;
rhoz     = 0.97;     //TFP autocorr. from linearly detrended Solow residual
Lambda_x= 0.0055;    //2.2% output growth per year

model;
#Rbar     = exp(Lambda_x)/beta;
1/exp(c_tilde)=beta/exp(mu_tilde(+1))*1/exp(c_tilde(+1))*(alpha*exp(z(+1))*(
    exp(k_tilde)/exp(mu_tilde(+1)))^(alpha-1)+(1-delta));
1/exp(c_tilde)=beta/exp(mu_tilde(+1))*1/exp(c_tilde(+1))*(exp(R_tilde)/exp(
    Pi_tilde(+1)));
exp(z)*(exp(k_tilde(-1))/exp(mu_tilde))^alpha=exp(c_tilde)+exp(k_tilde)-(1-
    delta)*exp(k_tilde(-1))/exp(mu_tilde);
exp(y_tilde)=exp(z)*(exp(k_tilde(-1))/exp(mu_tilde))^alpha;
exp(R_tilde)/Rbar=(exp(Pi_tilde)/Pibar)^phi_pi;
z=rhoz*z(-1)+eps_z;
mu_tilde=Lambda_x+eps_x;
end;
```

```
steady_state_model;
mu_tilde=Lambda_x;
k_tilde=log(exp(mu_tilde)*((exp(mu_tilde)/beta-(1-delta))/alpha)^(1/(alpha-1))
    );
y_tilde=log((exp(k_tilde)/exp(mu_tilde))^alpha);
c_tilde=log(exp(y_tilde)-(1-(1-delta)/exp(mu_tilde))*exp(k_tilde));
R_tilde=log(exp(mu_tilde)/beta);
Pi_tilde=log(Pibar);
z=0;
end;


shocks;
var eps_z=0.0068^2;
var eps_x=0.005^2; //just some number
end;

steady;
check;

stoch_simul(order=1,irf=20);
```

The goal of the observation equation again is to match empirical data to variables like y_t.

5.2 Integrated Variables like Output

As detailed above, detrended output y_t is given by the division of model output per capita Y_t by its trend X_t:[30]

$$y_t = \frac{Y_t}{X_t} \tag{61}$$

This implies that observed output per capita $Y_{pc,t}^{data}$ corresponds to the model variables

$$Y_{pc,t}^{data} = Y_t = y_t X_t \tag{62}$$

This equation cannot be entered into Dynare as the X_t multiplying our intensive form model variable y_t is not a stationary variable. Only its growth rate in the detrended law of motion (59) is. In order to get rid of X_t, we can simply work with growth rates. Take log differences of Y_t to get percentage growth rates from the previous quarter and use them as observables

[30] Given that we are already considering variables in per capita terms, y_t effectively is the familiar output in intensive form $\frac{Y_t}{N_t X_t}$ from the Solow model.

y_t^{obs}:[31]

$$
\begin{aligned}
y_t^{obs} &= \log(Y_{pc,t}^{data}) - \log(Y_{pc,t-1}^{data}) && (63)\\
&= \log(Y_t) - \log(Y_{t-1}) \\
&\overset{(62)}{=} \log(y_t) + \log(X_t) - [\log(y_{t-1}) + \log(X_{t-1})] \\
&\overset{(59)}{=} \log(y_t) - \log(y_{t-1}) + \log(\mu_t) \\
&= \tilde{y}_t - \tilde{y}_{t-1} + \tilde{\mu}_t && (64)
\end{aligned}
$$

where μ_t is the quarterly gross growth rate of labor augmenting technology. y_t^{obs} is clearly stationary as both detrended output y_t and the growth rate μ_t are stationary. Moreover, y_t^{obs} exactly corresponds to quarterly GDP per capita growth rates. Thus, we have found our observation equation for output. We use quarterly growth rates as in equation (63) in the datafile and name them y^{obs}. Figure 11 shows the resulting data series. Then we use equation (64) as the observation equation.

We saw that output per capita Y_t grows at rate Λ_y in steady state. Thus, the growth rate y_t^{obs} will have mean/steady state Λ_y, which is the long-run growth rate of the economy. This can be seen by considering equation (64) in steady state:

$$\bar{y}^{obs} = \bar{\tilde{y}} - \bar{\tilde{y}} + \bar{\tilde{\mu}} \overset{(59)}{=} \Lambda_x \,, \tag{65}$$

Effectively, the mean growth rate $\bar{\tilde{\mu}} = \Lambda_x$ is added to the growth rate of the stationary variables. If you use data in first differences without demeaning (as is the case for the left-hand side of equation (64)), this constant has to be accounted for by adding $\tilde{\mu}$ on the right-hand side.[32] If you erroneously forget about this (on average) positive growth rate, you will force the shocks to account for a positive mean in the growth rates (see Remark 13).

Summarizing, the example of output suggests a simple way of dealing with integrated $I(1)$

[31] I am not covering the case of an already log-linearized model with a stochastic growth trend, where variables $\hat{y}_t$ were percentage deviations from steady state. However, an extension to this case is straightforward. Consider equation (64) and add zero in the form of $\log(\bar{y}) - \log(\bar{y})$ and $\log(\bar{\mu}) - \Lambda_z$ to get

$$y_t^{obs} = \log(y_t) - \log(\bar{y}) - (\log(y_{t-1}) - \log(\bar{y})) + (\log(\mu_t) - \log(\bar{\mu})) + \Lambda_z = \hat{y}_t - \hat{y}_{t-1} + \widehat{\log(\mu_t)} + \Lambda_z$$

This provides the corresponding observation equation that only uses variables in percentage deviations and deep parameters of the model (i.e. Λ_z).

[32] Alternatively, you could have used demeaned growth rates, i.e. effectively subtracting $\bar{\mu}$ from both sides:

$$y_t^{obs} = \log(Y_{pc,t}^{data}) - \log(Y_{pc,t-1}^{data}) - \text{mean}(\log(Y_{pc,t}^{data}) - \log(Y_{pc,t-1}^{data})) = \log(y_t) - \log(y_{t-1}) = \tilde{y}_t - \tilde{y}_{t-1} + \tilde{\mu}_t - \Lambda_z$$

This of course comes at the loss of sample information about the mean.

variables in the context of models with explicitly specified trend: enter them in growth rates into the model and link them to the model variables by means of an observation equation of the type shown in equation (64).

5.3 Stationary Variables like Inflation and the Interest Rate

The previous section has shown how to deal with integrated variables. But there are also stationary $I(0)$ variables in the model like Π_t or R_t. Here, the model builder faces a choice between two possibilities, which are covered in turn.

5.3.1 Log Deviation from Mean

There is still the possibility to use an observation equation as was the case of the model abstracting from trend growth in Section 4, i.e. use log deviations from the mean. In this case, for inflation we have the familiar data transformation

$$\Pi_t^{obs} = \log\left(\Pi_t^{data}\right) \tag{50}$$

with the corresponding observation equation

$$\Pi_t^{obs} = \tilde{\Pi}_t \tag{51}$$

The data transformation for the nominal interest rate is also already known:

$$R_t^{obs} = \log\left(1 + \frac{R_t^{data}}{4 \times 100}\right) \tag{52}$$

as is the corresponding observation equation

$$R_t^{obs} = \tilde{R}_t \,. \tag{53}$$

The resulting listing is shown in Listing 10.

Listing 10: Observation Equation in Nonlinear Model for Log-Linearization using first differences for non-stationary variables

```
var mu_tilde Pi_tilde Pi_obs R_tilde R_obs ...;
parameter Pi_bar beta Lambda_x...;
model;
#Rbar = exp(Lambda_x)/beta;


 ...
```

```
y_obs=y_tilde-y_tilde(-1)+mu_tilde; //matched to non-demeaned growth rate
Pi_obs= Pi_tilde; //matched to log level
R_obs = R_tilde;  //matched to log level
end;
```

5.3.2 Log Growth Rates

Alternatively, one could use first differences also for the stationary variables. Of course this comes at the cost of losing all information about the level of inflation and interest rates (or alternatively the benefit of not being required to take a stance on the mean). Here, we would have

$$\Pi_t^{obs} = \log\left(\Pi_t^{data}\right) - \log\left(\Pi_{t-1}^{data}\right) \tag{66}$$

with the corresponding observation equation[33]

$$\Pi_t^{obs} = \tilde{\Pi}_t - \tilde{\Pi}_{t-1} \tag{67}$$

Similarly, the data transformation for the nominal interest rate is given by:[34]

$$R_t^{obs} = \log\left(1 + \frac{R_t^{data}}{4 \times 100}\right) - \log\left(1 + \frac{R_{t-1}^{data}}{4 \times 100}\right) \tag{68}$$

The corresponding observation equation immediately follows:

$$R_t^{obs} = \tilde{R}_t - \tilde{R}_{t-1} \ . \tag{69}$$

The resulting Listing is shown in Listing 11.

Listing 11: Observation Equation in Nonlinear Model for Log-Linearization using (demeaned) first differences for all variables

```
var mu_tilde Pi_tilde Pi_obs R_tilde R_obs ...;
parameter Pi_bar beta Lambda_x...;
model;
```

[33]For a log-linear model, this implies an observation equation of the type

$$\Pi_t^{obs} = \tilde{\Pi}_t - \bar{\Pi} - (\tilde{\Pi}_{t-1} - \bar{\Pi}) = \hat{\Pi}_t - \hat{\Pi}_{t-1}$$

[34]Again, in a log-linear model, this implies an observation equation of the type

$$R_t^{obs} = \tilde{R}_t - \bar{R} - (\tilde{R}_{t-1} - \bar{R}) = \hat{R}_t - \hat{R}_{t-1}$$

```
#Rbar = exp(Lambda_x)/beta;

...
y_obs=y_tilde-y_tilde(-1); //matched to demeaned growth rate
Pi_obs= Pi_tilde-Pi_tilde(-1); //matched to growth rate
R_obs = R_tilde-R_tilde(-1);
end;
```

Remark 27 (Demeaning growth rates of stationary variables)
In Listing 11, growth rates are used for the stationary variables inflation and interest rate. Theoretically, the population mean of these growth rates should be 0. However, this is often not the case for their sample averages. Thus, the researcher is faced with the choice whether to demean the data growth rates. Unless one has good reasons, one should abstain from demeaning them as the sample mean may contain important information. For example, a mean inflation growth rate bigger than 0 might indicate that over the sample considered there were more inflationary than disinflationary shocks. This may then help explain why output growth behaved the way it did. In contrast, demeaning the inflation growth rate and then entering this demeaned growth rate into the model effectively tells the model that there have been as many positive as negative shocks.

6 Dealing with Measurement Error

Up to this point, we have neglected the measurement errors affecting the observation equation (17). In general, there are three reasons for assuming the presence of measurement error in economics - one economic and two technical ones. First, some time series like wages and hours worked are notoriously noisy and poorly measured (see e.g. Justiniano, Primiceri, and Tambalotti 2013). Allowing for measurement error when observing these series can be economically justified by the data quality. A special case of this is when only a proxy is observed (see e.g. Justiniano, Primiceri, and Tambalotti 2011). Second, allowing for measurement error may be a way to account for model misspecification when the data violates the cross-equation restrictions implied by the model (see Del Negro and Schorfheide 2009; Sargent 1989). Finally, adding measurement error may be a way to circumvent stochastic singularity of the model (see e.g. Schmitt-Grohé and Uribe 2012). Stochastic singularity arises when there are more observables than shocks in the model or if the model equations imply an exact linear combination between (a subset of) the observables. The most common example of the latter is observing all components of the resource constraint of the model, i.e. output, consumption, and investment in our model. In this case, only one particular combination of shocks is consistent with the particular linear combination of observables and the forecast error covariance matrix will be singular. Roughly speaking, every exact combination of shocks has mass zero given our continuous distribution for shocks. Thus, the likelihood of observing such a linear combination is exactly zero, rendering estimation impossible. In contrast, assuming that e.g. output is only observed with error restores a full distribution and solves this stochastic singularity problem.

Remark 28 (Number of observables vs. number of shocks)
As discussed in Section 6, it is not possible to estimate a model with fewer shocks than observables, because stochastic singularity arises. Thus, many prominent papers in the literature use as many shocks as observables (e.g. Rabanal and Rubio-Ramírez 2005; Smets and Wouters 2007). But having more shocks than observables is also not uncommon (e.g. Ireland 2004; Schmitt-Grohé and Uribe 2012) and poses no problem as long as the parameters you are estimating are still identified. A sufficient condition is that any two shocks affect some moments of the observed data differently. In practice this often means that two shocks must not enter the same equations always as a linear combination. If in doubt, it is always advisable to use Dynare's `identification`-command to check whether the parameters are identifiable.

For whatever reason you want to add estimation error to your model, there are two ways to achieve this.

6.1 Measurement Errors as a Special Case of Exogenous variables

In general, there is no reason to treat measurement error as a special type of shock. As shown in the state-space representation of our solution, (16)-(17), it enters only the observation equation (17). But as detailed in Section 1.3, the observation equation is only implicitly specified in Dynare by providing a relation between the observed data and the model variables (not necessarily only the states as in equation (17)). This implies that one can treat measurement error as a special case of structural shocks that only affect the observed variables but not the state variables. When computing the model solution, they will thus only enter the observation equation. An example listing for a log-linearized model with measurement error in output is shown in Listing 12. As can be seen, the structural shock to technology `eps_z` is exactly treated the way the measurement error `eps_y_ME` is. The distinction from true structural shocks derives from the fact that the measurement error only enters our implicit observation equation and nowhere else in the model. Thereby, any shock `eps_y_ME` will only affect the observed variable `y_obs`, but no other variables in the model.

Listing 12: Observation Equations for Log-Linearized Model with Structural Shocks as Measurement Error

```
var y_hat R_hat Pi_hat... Pi_obs R_obs y_obs;
var_exo eps_z,eps_y_ME;

model;
...
Pi_obs=Pi_hat;
R_obs=R_hat;
y_obs=y_hat+eps_y_ME;
end;

estimated_params;
stderr eps_z,...;
stderr eps_y_ME,...;
end;

varobs Pi_obs R_obs y_obs;
```

6.2 Measurement Errors Using Dynare Capabilities

The second way is using Dynare's built-in capabilities for specifying measurement error in the observation equation. In this case, measurement errors are not treated as a special case of structural shocks, but rather are really the additive, serially uncorrelated shocks to the observation equation as seen in equation (17). The way to indicate to Dynare this form of measurement error is directly specifying the standard error on observed variables. For our example of measurement error on output, instead of defining an exogenous shock `eps_y_ME` and estimating its standard deviation, we specify immediately the standard error associated with the measurement error on `y_obs` using `stderr y_obs,...;` in the `estim_params`-block. The resulting mod-file, which is equivalent to the one in Listing 12 is shown in Listing 13.

Listing 13: Observation Equations for Log-Linearized Model with Measurement Error, Using Dynare's built-in Commands

```
var y_hat R_hat Pi_hat... Pi_obs R_obs y_obs;
var_exo eps_z;
```

```
model;
...
Pi_obs=Pi_hat;
R_obs=R_hat;
y_obs=y_hat;
end;

estimated_params;
stderr eps_z,...;
stderr y_obs,...;
end;

varobs Pi_obs R_obs y_obs;
```

6.3 Which Way of Specifying Measurement Error is Better?

As always in economics, the answer is: it depends. On the one hand, using built-in Dynare capabilities is typically easier and allows to introduce contemporaneously correlated measurement error by using commands as shown in Listing 14.

Listing 14: Observation Equations for Log-Linearized Model with Correlated Measurement Error, Using Dynare's built-in Commands

```
var y_hat R_hat Pi_hat... Pi_obs R_obs y_obs;
var_exo eps_z;

model;
...
Pi_obs=Pi_hat;
R_obs=R_hat;
y_obs=y_hat;
end;

estimated_params;
stderr eps_z,...;
stderr y_obs,...;
stderr Pi_obs,...;
corr Pi_obs, y_obs,...;
end;

varobs Pi_obs R_obs y_obs;
```

When trying to do this without the built-in Dynare capabilities, you need to add an additional joint shock on top of the exogenous shocks to `Pi_obs` and `y_obs`, which is cumbersome.

On the other hand, treating measurement error as a special case of structural shocks allows for more flexibility compared to the built-in Dynare capabilities. For example, Dynare's built-in capabilities only allow for measurement error that is additive to the observed data in its final transformed form entered to the model, y_t^{obs}, and serially uncorrelated.[35] This is the most common way to specify measurement error in the literature and also the way I specified it in Listings 12 and 13. But there are cases where one might want to allow for more complicated processes. Treatment of measurement error similar to any other structural shock easily allows this! Any process you could specify for structural shocks would be also possible for measurement error. You could for example allow for serial correlation or multiplicative measurement error. However, in the latter case, you have to keep in mind that the equations will be linearized by Dynare, which will restore some form of linearity.

7 Time Aggregation

The way of writing down DSGE model setups with variables having just a time index like t often masks the intricate timing structure of economic variables in practice that needs to be captured within our models. The time index t of discrete time models represents a time period rather than a point in time. For example, if the model is quarterly, the period $t = 1$ can mean the first quarter of year 2013, i.e. the time period between January 1st, 2013 and March 31st, 2013. There are three different types of variables that must be distinguished according to their timing structure (see Figure 16 for a stylized timeline):

1. Variables that measure economic concepts *within* a certain time period, i.e. between the beginning and end of a period like a quarter. Examples are flow variables like GDP.

2. Variables that are measured *at* a certain point in time, e.g. at the end of a quarter. Examples are stock variables like bonds or capital.

3. Variables that are measured *at* a certain point in time, but measure an economic concept *between* points in time. Typical examples are interest and inflation rates, e.g. the interest rate at January 1st, 2013 for a loan until April 1st, 2013. The major conceptual difference to flow variables is that flow variables measure something within a time period

[35] This creates its own consistency issues: is it actual GDP data Y_t^{data} that is measured with error or the transformed/filtered data y_t^{obs}. And if it is Y_t^{data} that is measured with error, what does this imply for the (functional form of the) measurement error contained in the filtered data? See also Canova and Ferroni (2011).



Figure 16: Simulated output series from the mod-file presented in Listing 1.

while interest and inflation rates typically link time points.[36] Of course, one could also define intra-period interest rates (e.g. the interest rate for a loan in 2013Q1, i.e. between January 1st, 2013 and March 31st, 2013), but macro-models often assume that within periods there is no interest paid (e.g. Jermann and Quadrini 2012).

Correctly accounting for the different timing structure when matching data and model variables is important. This section discusses some of the issues arising in this context, with a particular focus on time aggregation.[37]

7.1 Flow Variables

Flow variables like output or investment are measured between two points in time. Hence, the length of a model period, i.e. the frequency of the model matters. If the length of a model period doubles, the corresponding flow variable will also double. While this seems straightforward, there is an important difference between non-linear and log-linearized models due to the different scaling introduced by the transformation in percentage deviations.

It is the *level* of flow variables like GDP and investment, which is used in non-linear models, that scales with the model frequency. Denote months with subscript $m = m1, m2, \ldots, m12$ and quarters with subscript $q = q1, \ldots, q4$. Consider e.g. quarterly output in the first quarter $Y_q 1$. It is the sum of the monthly outputs in the first three months:

$$Y_{q1} = Y_{m1} + Y_{m2} + Y_{m3} \, . \tag{70}$$

For example, if GDP per month in steady state is 100 apples, 300 apples will be produced per quarter.

[36]The conceptual difference to flow variables is only present in discrete time. In continuous time, both are essentially the same as flow variables are the integral over instantaneous flows and interest rates between time points are the integral over instantaneous interest rates.

[37]The problem of time aggregation was first encountered in the context of interest rates in Section 3.4. Because empirical interest rates were quoted in annual frequency, but our model was set up in quarterly frequency, we had to divide the net interest rates by four/invert the geometric average.

However, things are different in a loglinearized model. Performing a log-linearization of equation (70) around the deterministic steady state yields:

$$\bar{Y}_q \hat{Y}_{q1} = \bar{Y}_m \hat{Y}_{m1} + \bar{Y}_m \hat{Y}_{m2} + \bar{Y}_m \hat{Y}_{m3} \,, \tag{71}$$

where bars denote steady state values and hats percentage deviations from steady state. Divide by $\bar{Y}_q = 3\bar{Y}_m$ to obtain quarterly percentage deviations:

$$\hat{Y}_{q1} = \frac{\bar{Y}_m}{\bar{Y}_q}\hat{Y}_{m1} + \frac{\bar{Y}_m}{\bar{Y}_q}\hat{Y}_{m2} + \frac{\bar{Y}_m}{\bar{Y}_q}\hat{Y}_{m3} = \frac{1}{3}\left(\hat{Y}_{m1} + \hat{Y}_{m2} + \hat{Y}_{m3}\right) \,. \tag{72}$$

Thus, to aggregate the percentage deviations from steady state of monthly data to quarterly frequency, the mean must be used, not the sum as is the case for levels. Staying with our example, assume that a mini-recession hits and only 99 apples are produced in the first month and 100 in the other two. The absolute deviation of GDP from its steady state is 1 apple. However, in percentage terms, the deviation is $299/300 - 1 = -1/3$ percent.

If you erroneously use the sum and not the mean, you will overestimate the movement of model variables compared to the data by a factor of 3 (see Born and Pfeifer 2014b, who document such an example).

7.2 Stock Variables

In contrast to flow variables, stock variables like capital or bonds are measured with their (average) value at a particular point in time. Hence, the length of a model period/the frequency of the model does not affect them in the same way as for flow variables. If the capital stock is 400 apples at any given point in time, it will be 400 apples whether the model is an annual or a quarterly one. If the capital stock varies over the intermediate time periods, again the mean is appropriate as was the case for loglinearized flow variables: the average annual capital stock is the mean of the quarterly capital stocks.

A critical issue with stock variables arises from the fact that models are often calibrated to target particular stock-flow ratios like the debt to GDP ratio. Here, the denominator, GDP, is affected by the model frequency, while the numerator is not. Not accounting for this difference when the data is quoted at annual frequency and the model is at quarterly frequency (or vice versa) is a common source of error. For example, the US debt to GDP ratio from 1970Q1 to 2012Q2 was 37.5% of annual output. If the model is calibrated to quarterly frequency, quarterly GDP is only one fourth of annual GDP, implying that the quarterly debt to GDP ratio in the data is $4 \times 37.5\% = 150\%$. The quarterly model must then target this 150% of quarterly GDP.

For stocks, another critical issue is the timing. Dynare uses the end of period stock timing convention. That is, k_t and b_t represent the stock of capital and bonds at the end of period t, respectively. Using stocks as observables if often challenging for two reasons. First, it is hard to find data that follows the timing convention used or, in many cases, to even find any information on the timing of a measured stock. For example, the US Multifactor Productivity (MFP) tables (http://www.bls.gov/mfp/) provide information on the US capital stock at the annual level. Thus, one would typically need the end of year capital stocks, but the data in the MFP tables is linearly interpolated to mid-year data.[38] Second, the stock data is often measured differently from the other (flow) data in the model, making it inconsistent with the flow data measuring additions to and subtraction from this stock. For example, US Federal Debt: Total Public Debt (FRED: GFDEBTN) is actually available as an end of the period quarterly stock. But in the context of models of fiscal behavior one may need a series that is consistent with NIPA's net borrowing definition. That is why e.g. Leeper, Plante, and Traum (2010) reconstruct a debt series from NIPA deficit concepts.[39] In such cases, it might be preferable to abstain from using stock data.

7.3 Mixed Frequency without Explicit Growth Trend

In some instances, one would like to estimate a model at a relatively high frequency like monthly, but some data are only available at a lower frequency like e.g. quarterly frequency. A typical example are models of the labor market where labor market flows are available at monthly frequency (e.g. Shimer 2005), but GDP is only available at quarterly frequency. When monthly industrial production is not an alternative to GDP, the only option is to estimate the model using data at different frequencies. Doing so in Dynare is straightforward due to the implemented Kalman filter that is capable of dealing with missing data values. The underlying idea is to specify the theoretical monthly frequency construct that is underlying the observed quarterly data for all months, $m = m1, \ldots, m12$ and treat only the values after $m3, m6, m9$, and $m12$ as observed. All other realizations of the theoretical monthly construct are set to be missing. As outlined in detail in Durbin and Koopman (2012, Chapter 4.10), the Kalman filter will then treat those observations as missing values and will try to infer their values given the observables.

[38] See Bureau of Labor Statistics (1983, p.41): "The year-end stocks are then averaged with the previous year-end stock to estimate the services contributed by a given type of asset during the year". I am greatly indebted to Steven Rosenthal for providing this source.

[39] If this is not a constraint, the Dallas FED provides an end-of-month market value of debt series at http://www.dallasfed.org/research/econdata/govdebt.cfm that follows the work of Cox and Hirschhorn (1983).

Remark 29 (Low-Frequency Data and Observational Equivalence)
It should be noted that high frequency data can sometimes be essential for correctly estimating structural parameters, because data at lower frequency might only restrict a non-invertible mapping of the parameter(s) of interest. Put differently, with missing high-frequency observations, different parameter combinations can become observationally equivalent. Consider the example of an AR1-process that is only observed every second period. Plugging in twice, we get $y_t = \rho^2 y_{t-2} + \rho\varepsilon_{t-1} + \varepsilon_t$ where $\rho > 0$ is to be estimated. By just observing y_t and y_{t-2} there is no way to distinguish the two roots of ρ^2 at $\pm\rho$ as they are observationally equivalent.

For simple examples like the one above it is easy to spot the problems. However, for more complicated models the identifiability of the parameters is hard to establish, because standard identification tests (e.g. Iskrev 2010; Ratto and Iskrev 2011) do not take missing data into account. A good strategy thus is to start estimation at different starting values and see whether the estimation converges to the same parameter set.

There are no easy remedies for the problem of observational equivalence with missing data. Sometimes, the multiplicity can be eliminated by a priori ruling out parts of the parameter space by e.g. specifying that $\rho > 0$ because oscillating behavior is hardly economically sensible. Another potential solution is to add an additional observed variable, say z_t, that restricts the value of ρ via its impact on z_t.

I will in the following use a monthly version of our classical monetary economy as an illustrative example. The assumption is that we want to estimate the model at monthly frequency and have two observed variables: monthly interest rates and quarterly GDP. Denote months with subscript $m = m1, m2, \ldots, m12$ and quarters with subscript $q = q1, \ldots, q4$. Estimating this model using this mixed-frequency data requires the three steps outlined in the following.

First, set up the model at the frequency of the most high-frequent variable you want to use as an observable. In our illustrative example, this would be at monthly frequency. Listing 15 shows the CME at monthly frequency. Note that the interest rates, the depreciation rate, and the autoregressive coefficients were adjusted to reflect the monthly frequency.

Listing 15: Basic RBC Classical Monetary Economy Model at Monthly Frequency

```
var y_tilde c_tilde k_tilde z R_tilde Pi_tilde y_tilde_quarterly;
varexo eps_z;

parameters beta delta alpha rhoz phi_pi Pibar;
```

```
alpha   = 0.33;       // capital share
delta   = 0.025/3;     //deprecation rate
beta    = 0.99^(1/3);      //discount factor
Pibar   = 1;
phi_pi  = 1.5;
rhoz    = 0.97^(1/3);      //TFP autocorr. from linearly detrended Solow residual

model;
#Rbar     = 1/beta;
1/exp(c_tilde)=beta*1/exp(c_tilde(+1))*(alpha*exp(z(+1))*exp(k_tilde)^(alpha-1)+(1-delta));
1/exp(c_tilde)=beta*1/exp(c_tilde(+1))*(exp(R_tilde)/exp(Pi_tilde(+1)));
exp(z)*exp(k_tilde(-1))^alpha=exp(c_tilde)+exp(k_tilde)-(1-delta)*exp(k_tilde(-1));
exp(y_tilde)=exp(z)*exp(k_tilde(-1))^alpha;
exp(R_tilde)/Rbar=(exp(Pi_tilde)/Pibar)^phi_pi;
z=rhoz*z(-1)+eps_z;
exp(y_tilde_quarterly)=exp(y_tilde)+exp(y_tilde(-1))+exp(y_tilde(-2));
end;

steady_state_model;
k_tilde=log(((1/beta-(1-delta))/alpha)^(1/(alpha-1)));
y_tilde=log(exp(k_tilde)^alpha);
c_tilde=log(exp(y_tilde)-delta*exp(k_tilde));
R_tilde=log(1/beta);
Pi_tilde=log(Pibar);
z=0;
y_tilde_quarterly=log(3*exp(y_tilde));
end;


shocks;
var eps_z=0.0068^2;
end;

steady;
check;

stoch_simul(order=1,irf=20,periods=750);

varobs y_tilde R_tilde;
```

Second, define an observation equation that at each point in time m defines the theoretical relation between the high-frequency model variables and the low-frequency observed variables. In our illustrative example, we are observing quarterly GDP, which is just the sum of monthly GDPs in the previous three months, i.e.

$$Y_{q1} = Y_{m1} + Y_{m2} + Y_{m3} \; . \tag{70}$$

While we only observe the sum of the previous three months' GDP after months $3, 6, 9$, and 12, this theoretical construct is valid at all points in time m. If we would be able to observe "quarterly GDP in April", it would be the sum of the GDPs in February, March, and April. Line 21 of Listing 15 shows the definition of quarterly output as the sum of monthly output:
exp(y_tilde_quarterly)=exp(y_tilde)+exp(y_tilde(−1))+exp(y_tilde(−2));
This is the variable we observe, but only every third month, when it is equal to quarterly GDP.

Finally, provide the observed data at monthly frequency, setting the unobserved monthly values for the quarterly variables to *NaN*. As discussed in the previous point, we only observe quarterly GDP after months 3,6,9, and 12 while the values for months 1,2,4,5,7,8,10, and 11 are missing. The result of this is shown in Figure 17. For the more high-frequent interest rate, values are observed every month, while for quarterly GDP only values at the end of each quarter are observed.

7.4 Mixed Frequency with Explicit Growth Trend

In this section, in contrast to the previous one, we consider time aggregation in the context of an explicitly defined growth trend. A common economic problem, often encountered when working with emerging economy data, is the total or partial non-availability of quarterly national accounts. Quarterly GDP figures might only be available for the last 10 years or quarterly consumption numbers might not be available at all. In this case, one might want to combine available quarterly data with more easily available annual data. The model itself needs to be set up in quarterly frequency, i.e. t denotes a quarter. The observation equation needed is one linking the observed trending annual GDP in the data, $Y_{ann,t}^{data}$, to the stationary quarterly concept of GDP y_t and trend growth μ_t.

In every quarter t, we can define trending annual GDP, $Y_{ann,t}$, as the sum of the previous four quarters' trending GDP, Y_t:

$$Y_{ann,t} = Y_t + Y_{t-1} + Y_{t-2} + Y_{t-3} \; , \tag{73}$$



	A	B	C	D	E	F	G	H	I	J	K
1	R_tilde	c_tilde_quarterly									
2	0.003467899										
3	0.003866377										
4	0.003357164	2.478338673									
5	0.003545678										
6	0.003609909										
7	0.003315997	2.477899937									
8	0.003221666										
9	0.003300008										
10	0.004085067	2.478502977									
11	0.004671119										
12	0.004338282										
13	0.004974918	2.492890901									
14	0.005087665										
15	0.00502412										
16	0.005132267	2.500821343									
17	0.005035481										
18	0.004958607										
19	0.005236955	2.504328637									
20	0.005489462										
21	0.00573623										

Figure 17: Datafile for mixed-frequency estimation, using missing values for unobserved time periods.

where $Y_{ann,t}$ inherits trend X_t and can be decomposed into a stationary component $y_{ann,t}$ and the trend:

$$Y_{ann,t} = y_{ann,t} X_t \; . \tag{74}$$

In the actual data, we observe the sum of the quarterly GDPs, $Y^{data}_{ann,t}$, only every fourth period. The setup on the data side therefore mirrors what has been shown in the previous section. What is different in this section is that we consider the growth trend explicitly and are therefore linking the model to the quarterly growth rate of annual GDP, $\Delta Y^{obs}_{ann,t}$:

$$\Delta Y^{obs}_{ann,t} = \log Y^{data}_{ann,t} - \log Y^{data}_{ann,t-4} \tag{75}$$

This growth rate of annual GDP is given by the log difference of today's measurement of annual GDP (comprising the quarters $t, t-1, t-2$, and $t-3$), $Y^{data}_{ann,t}$m and the measurement of annual GDP from time $t-4$ (comprising $t-4, t-5, t-6$, and $t-7$), $Y^{data}_{ann,t}$. This growth rate is again only observed every fourth quarter, i.e. at the end of each year. As before, the output growth rate can be linked to the model variables:

$$\Delta Y^{obs}_{ann,t} = \log Y^{data}_{ann,t} - \log Y^{data}_{ann,t-4}$$

$$
\begin{aligned}
&= \log Y_{ann,t} - \log Y_{ann,t-4} \\
&\overset{(74)}{=} \log\left(y_{ann,t} X_t\right) - \log\left(y_{ann,t-4} X_{t-4}\right) \\
&= \hat{y}_{ann,t} - \hat{y}_{ann,t-4} + \log\left(\frac{X_t}{X_{t-4}}\right) \\
&= \hat{y}_{ann,t} - \hat{y}_{ann,t-4} + \hat{\mu}_t + \hat{\mu}_{t-1} + \hat{\mu}_{t-2} + \hat{\mu}_{t-3} \qquad (76)
\end{aligned}
$$

The third line makes use of the fact that annual GDP the year before inherits trend X_{t-4}. The fourth line adds $0 = \log y_{ann} - \log y_{ann}$ and uses the definition of percentage deviations from trend

$$
\hat{y}_{ann,t} \equiv \log y_{ann,t} - \log y_{ann}\,. \qquad (77)
$$

The fifth line of equation (76) uses the definition of the trend growth rate $\hat{\mu}_t = \log X_t/X_{t-1}$.

Equation (76) is our desired observation equation. However, to make it operational and be able to add it to the model, we still need to define $\hat{y}_{ann,t}$. Making use of the definition of trending variables as the product of detrended variables times their growth trend, the definition of trending annual GDP in, equation (73), can be written as

$$
y_{ann,t} X_t = y_t X_t + y_{t-1} X_{t-1} + y_{t-2} X_{t-2} + y_{t-3} X_{t-3}\,. \qquad (78)
$$

Dividing by X_t and making use of the definition of the growth rate $\mu_t = X_t/X_{t-1}$ yields

$$
y_{ann,t} = y_t + y_{t-1}\mu_{t-1}^{-1} + y_{t-2}\mu_{t-1}^{-1}\mu_{t-2}^{-1} + y_{t-3}\mu_{t-1}^{-1}\mu_{t-2}^{-1}\mu_{t-3}^{-1} \qquad (79)
$$

Linearizing this equation around the steady state of

$$
y_{ann} = y\left(1 + \frac{1}{\mu} + \frac{1}{\mu^2} + \frac{1}{\mu^3}\right) \qquad (80)
$$

results in

$$
\begin{aligned}
y_{ann}\hat{y}_{ann,t} =& y\hat{y}_t + \frac{y}{\mu}\left(\hat{y}_{t-1} - \hat{\mu}_{t-1}\right) \\
&+ \frac{y}{\mu^2}\left(\hat{y}_{t-2} - \hat{\mu}_{t-1} - \hat{\mu}_{t-2}\right) \qquad (81) \\
&+ \frac{y}{\mu^3}\left(\hat{y}_{t-3} - \hat{\mu}_{t-1} - \hat{\mu}_{t-2} - \hat{\mu}_{t-3}\right)
\end{aligned}
$$

This can be solved for $\hat{y}_{ann,t}$, using the steady state definition (80) :

$$\begin{aligned}\hat{y}_{ann,t} =&\left(1+\frac{1}{\mu}+\frac{1}{\mu^2}+\frac{1}{\mu^3}\right)^{-1}\hat{y}_t+\frac{1}{\mu}\left(1+\frac{1}{\mu}+\frac{1}{\mu^2}+\frac{1}{\mu^3}\right)^{-1}(\hat{y}_{t-1}-\hat{\mu}_{t-1})\\&+\frac{1}{\mu^2}\left(1+\frac{1}{\mu}+\frac{1}{\mu^2}+\frac{1}{\mu^3}\right)^{-1}(\hat{y}_{t-2}-\hat{\mu}_{t-1}-\hat{\mu}_{t-2})\\&+\frac{1}{\mu^3}\left(1+\frac{1}{\mu}+\frac{1}{\mu^2}+\frac{1}{\mu^3}\right)^{-1}(\hat{y}_{t-3}-\hat{\mu}_{t-1}-\hat{\mu}_{t-2}-\hat{\mu}_{t-3})\end{aligned} \tag{82}$$

This definition can completes the implementation of equation (76).

Remark 30 (Time Aggregation and Labor Market Flows)
For labor market flows, things are more complicated as the correct model setup depends on the actual definition of the measured variables, which may differ across countries. Consider for example a US labor market model that is set up in weekly frequency, but where only a monthly unemployment rate is observed. The question is how to map this stock of unemployed people to the model. The answer lies in the definition of the unemployment rate. In the US, it its measured in the CPS:

"Each month, highly trained and experienced Census Bureau employees contact the 60,000 eligible sample households and ask about the labor force activities (jobholding and job seeking) or non-labor force status of the members of these households during the survey reference week (usually the week that includes the 12th of the month)." (http://www.bls.gov/cps/cps_htgm.htm)

The unemployment rate reported for each month actually corresponds to the weekly rate observed for one particular week. The values for all other weeks are simply missing. Thus, the observation equation mapping the weekly unemployment rate in the model to the monthly rate in the data is just an identity mapping with the dataset containing the monthly value for the reference week and missing values for all other weeks.

However, this approach is only valid for the US. For other countries, the definition might be different and the observation equations needs to be adjusted accordingly. For example, the regular German unemployment statistics also uses a reference point (mid-month since 2005, end of month before this). In contrast, the International Labour Organization (ILO)-concept based numbers also published by the German "Bundesagentur für Arbeit" are derived from a survey and are a monthly average (see Hartmann and Riede 2005, (in German)).

Finally, an additional complication is introduced when the numerator and denominator of variables like the unemployment rate are measured from different sources/with different methods. If for example the labor force, which is the denominator in the unemployment rate, is not measured every month or is a monthly average while the stock of unemployed people is measured at a reference point, this needs to be taken into account in the observation equation.

7.5 Interest Rates

Net interest rates like flow variables scale with the time period for which the are considered, but are usually quoted as the annualized interest rate between two points in time. Ignoring compounding for the moment, the net interest rate you would have to pay for a two quarter loan is twice the interest rate you would have to pay for a one quarter loan. For example if the annual interest rate is 4%, a one quarter loan would require 1% interest and a two quarter loan 2%. Thus, for net interest, time aggregation is performed by summing interest rates up. It is important to keep in mind that the same is not true for gross interest rates, where the product would be appropriate.

A practical problem of matching observed interest rates to model variables is that the one period nominal interest rate between time t and $t+1$, $R_{t,t+1}$ represents the interest rate between the two points in time, t and $t+1$. But in the model, the index t usually does not represent a time point, but a time period like a quarter. Common practice thus is to operationalize $R_{t,t+1}$ as the average one-period ahead interest rate in period t. For example, Smets and Wouters (2007) use the Effective Federal Funds rate (geometrically) averaged over all trading days within a given quarter and then transform this average annual interest rate to a quarterly value as discussed in Section 3.4.

8 Higher order solutions

For higher orders, things become more complicated. In those cases, the mean of the data does not correspond to the deterministic steady state. Thus, subtracting the steady state is not sufficient to achieve mean 0 data. Rather, one would have to subtract the ergodic mean, which is usually unknown. Hence, one should use undemeaned data and match this to undemeaned model variables as was done in Listing 10.

An additional complication arises because the ergodic mean is a complicated function of the model parameters, which makes it usually impossible to use a single parameter to fix a particular mean. That implies in particular that the typical way of setting hours worked in linear models to 1/3 at the deterministic steady state by fixing the labor disutility parameter



Figure 18: Log of real GDP for Ecuador, available only for different periods with different base years

does not work at higher order, because a deterministic steady state of 1/3 does not correspond to an ergodic mean of 1/3 (neither in the model nor in the data). Ruge-Murcia (2012) and Born and Pfeifer (2014a) have thus estimated the steady state labor disutility parameter in order to match the implied model ergodic mean to the mean in the data. For that purpose they used Average Weekly Hours of Production and Nonsupervisory Employees: Manufacturing (FRED: AWHMAN) and normalized it by 5 days times 24 hours. This results in a level data series with a mean of about 1/3. The labor disutility parameter is then estimated in order to match the ergodic mean of the model to this data mean.

9 On Splicing and Seasonal Adjustment

When dealing with data of less developed countries one often faces the problem that no consistent long time series are available. Rather, time series with different base years are available for different periods of time. For example, Figure 18 shows that Ecuador provides real GDP data with base year 2000 for 1993Q1 to 2009Q3, while data with base year 2007 is available starting in 2001Q. Usually, the available data series for different periods do not only differ with respect to the base year, but often there are additional conceptual changes and improvements in data sources and data collection. This usually results in a level shift in the respective time series, as visible in the figure.

The linking of such heterogeneous time series is called "splicing". Conceptually, there are three different ways of dealing with this. The most common way is using growth rates of the older series to "backcast" the more recent time series to previous periods. This procedure will preserve the growth rates of time.

Additional complications are introduced when the underlying data for splicing is not yet seasonally adjusted at the source. Basically, two fundamental issues occur.

9.1 Seasonal Adjustment Before or After Splicing?

This first question is to whether to perform seasonal adjustment on the individual short time series that are then spliced together or whether to first splice together the seasonally unadjusted shorter series and then seasonally adjust the long time series. There are arguments to be made for both approaches. Ideally, a longer time series allows for a more reliable estimation and thus removal of a seasonal pattern, suggesting that splicing should be conducted before seasonal adjustment. However, this requires that the seasonal pattern in all series to be spliced together is the same, which is almost impossible to ascertain. The risk is then to remove an average seasonal pattern that will not fit any of the small time-series and will leave each them with seasonality. Think about two equally long time-series to be spliced together where the first series has a Christmas peak of 5 percent while in the second time-series, due to changes in the base year and in the concepts used, the Christmas peak is 15 percent. The estimated average seasonal pattern in the fourth quarter will be 10 percent. When subtracting this pattern from the spliced series, the first half will now have a "negative" seasonal pattern of -5 percent while in the second half a seasonal spike of 5 percent will still be present. Thus, when not being sure it seems advisable to first seasonally adjust the individual time series and then do the splicing.

9.2 Seasonal Adjustment and Additivity of Components

An additional issue derives from the fact that there is regularly a direct relationship between different non-seasonally adjusted time series. For example, GDP is the sum of its components consumption, investment, government spending, and net exports. When removing seasonal patterns, one would ideally like to preserve this relationship, which means that the seasonal components in all subaggregates of GDP should sum to 0.

One way of dealing with this problem is to seasonally adjust the subcomponents of GDP and then to compute the seasonally adjusted GDP as the sum of these components instead of using the directly seasonally adjusted GDP series.

References

Acemoglu, Daron (2008). *Introduction to Modern Economic Growth.* Princeton University Press.

Adjemian, Stéphane, Houtan Bastani, Fréderic Karamé, Michel Juillard, Junior Maih, Ferhat Mihoubi, George Perendia, Johannes Pfeifer, Marco Ratto, and Sébastien Villemot (2011). "Dynare: reference manual version 4". Dynare Working Papers 1. CEPREMAP.

An, Sungbae and Frank Schorfheide (2007). "Bayesian analysis of DSGE models". *Econometric Reviews 26* (2-4), 113–172.

Baxter, Marianne and Robert G. King (1999). "Measuring Business Cycles: Approximate Band-Pass Filters For Economic Time Series". *Review of Economics and Statistics 81* (4), 575–593.

Born, Benjamin, Alexandra Peter, and Johannes Pfeifer (2013). "Fiscal News and Macroeconomic Volatility". *Journal of Economic Dynamics and Control 37* (12), 2582–2601.

Born, Benjamin and Johannes Pfeifer (2014a). "Policy risk and the business cycle". *Journal of Monetary Economics 68*, 68–85.

——— (2014b). "Risk Matters: The Real Effects of Volatility Shocks: Comment". *American Economic Review 104* (12), 4231–4239.

Bureau of Labor Statistics (1983). "Appendix C. Capital Input and Capital and Labor Shares". *Trends in Multifactor Productivity, 1948-81*. Vol. 2178. BLS Bulletin.

Burnside, Craig (1998). "Detrending and business cycle facts: A comment". *Journal of Monetary Economics 41* (3), 513–532.

Canova, Fabio (1998a). "Detrending and business cycle facts". *Journal of Monetary Economics 41* (3), 475–512.

——— (1998b). "Detrending and business cycle facts: A user's guide". *Journal of Monetary Economics 41* (3), 533–540.

Canova, Fabio and Filippo Ferroni (2011). "Multiple filtering devices for the estimation of cyclical DSGE models". *Quantitative Economics 2* (1), 73–98.

Canova, Fabio and Luca Sala (2009). "Back to square one: Identification issues in DSGE models". *Journal of Monetary Economics 56* (4), 431–449.

Christiano, Lawrence J., Martin Eichenbaum, and Robert Vigfusson (2006). "Assessing Structural VARs". *NBER Macroeconomics Annual*. Ed. by Daron Acemoglu, Kenneth Rogoff, and Michael Woodford. National Bureau of Economic Research. Chap. 1, 1–106.

Clarida, Richard, Jordi Galí, and Mark Gertler (2000). "Monetary Policy Rules And Macroeconomic Stability: Evidence And Some Theory". *Quarterly Journal of Economics 115* (1), 147–180.

Cox, W. Michael and Eric Hirschhorn (1983). "The market value of U.S. government debt; Monthly, 1942-1980". *Journal of Monetary Economics 11* (2), 261–272.

Davig, Troy and Eric M. Leeper (2007). "Generalizing the Taylor Principle". *American Economic Review 97* (3), 607–635.

Del Negro, Marco and Frank Schorfheide (2009). “Monetary Policy Analysis with Potentially Misspecified Models”. *American Economic Review 99* (4), 1415–50.

Durbin, J. and Siem J. Koopman (2012). *Time Series Analysis by State Space Methods*. Second Revised Edition. Oxford: Oxford University Press.

Edge, Rochelle M., Refet S. Gürkaynak, and Burçin Kisacikoğlu (2013). “Judging the DSGE Model by Its Forecast”. Tech. rep.

Edge, Rochelle M., Thomas Laubach, and John C. Williams (2007). “Learning and shifts in long-run productivity growth”. *Journal of Monetary Economics 54* (8), 2421–2438.

Fernández-Villaverde, Jesús (2009). “The Econometrics of DSGE Models”. *SERIEs 1* (1-2), 3, 49.

Francis, Neville and Valerie A. Ramey (2009). “Measures of per Capita Hours and Their Implications for the Technology-Hours Debate”. *Journal of Money, Credit and Banking 41* (6), 1071–1097.

Galí, Jordi (2008). *Monetary Policy, Inflation and the Business Cycle*. Princeton University Press.

Gómez, Victor and Agustín Maravall (1996). “Programs TRAMO and SEATS, Instruction for User (Beta Version: september 1996)”. Banco de España Working Papers 9628. Banco de España.

Granger, Clive W. J. (1979). “Seasonality: Causation, Interpretation, and Implications”. *Seasonal Analysis of Economic Time Series*. Ed. by Arnold Zellner. NBER Chapters. National Bureau of Economic Research, 33–56.

Guerron-Quintana, Pablo A. (2010). “What you match does matter: the effects of data on DSGE estimation”. *Journal of Applied Econometrics 25* (5), 774–804.

Hartmann, Michael and Thomas Riede (2005). “Erwerbslosigkeit nach dem Labour-Force-Konzept – Arbeitslosigkeit nach dem Sozialgesetzbuch: Gemeinsamkeiten und Unterschiede”. *Wirtschaft und Statistik* (4/2005), 303–310.

Hodrick, Robert J. and Edward C. Prescott (1980). “Post-war U.S. business cycles: an empirical investigation”. Tech. rep. Carnegie-Mellon University, Pittsburgh.

Iacoviello, Matteo and Stefano Neri (2010). “Housing Market Spillovers: Evidence from an Estimated DSGE Model”. *American Economic Journal: Macroeconomics 2* (2), 125–64.

Ireland, Peter (2004). “Technology shocks in the New Keynesian model”. *Review of Economics and Statistics 86* (4), 923–936.

Ireland, Peter and Scott Schuh (2008). “Productivity and U.S. Macroeconomic Performance: Interpreting the Past and Predicting the Future with a Two-Sector Real Business Cycle Model”. *Review of Economic Dynamics 11* (3), 473–492.

Iskrev, Nikolay (2010). “Local identification in DSGE models”. *Journal of Monetary Economics 57*, 189–202.

——— (2011). “Evaluating the strength of identification in DSGE models. An a priori approach”. Tech. rep. Banco de Portugal.

Jermann, Urban and Vincenzo Quadrini (2012). “Macroeconomic effects of financial shocks”. *American Economic Review 102* (1), 238–71.

Jones, Charles I. (2002). “Using chain-weighted NIPA data”. *FRBSF Economic Letter* (22).

Justiniano, Alejandro, Giorgio E. Primiceri, and Andrea Tambalotti (2011). “Investment shocks and the relative price of investment”. *Review of Economic Dynamics 14* (1), 101–121.

——— (2013). “Is there a trade-off between inflation and output stabilization?” *American Economic Journal: Macroeconomics 5* (2), 1–31.

Kim, Jinill, Sunghyun Kim, Ernst Schaumburg, and Christopher A. Sims (2008). “Calculating and using second order accurate solutions of discrete time dynamic equilibrium models”. *Journal of Economic Dynamics and Control 32* (11), 3397–3414.

King, Robert G. and Sergio T. Rebelo (1999). “Resuscitating real business cycles”. *Handbook of Macroeconomics*. Ed. by John B. Taylor and Michael Woodford. Vol. 1. Elsevier, 927–1007.

Kliem, Martin and Alexander Kriwoluzky (2014). “Toward a Taylor rule for fiscal policy”. *Review of Economic Dynamics 17*, 294–302.

Kornfeld, Robert J., Brian C. Moyer, George M. Smith, David F. Sullivan, and Robert E. Yuskavage (2008). “Improving BEA’s accounts through flexible annual revisions”. *Survey of Current Business 88*, 29–32.

Leeper, Eric M., Michael Plante, and Nora Traum (2010). “Dynamics of Fiscal Financing in the United States”. *Journal of Econometrics 156* (2), 304–321.

Maravall, Agustń (1999). “Unobserved Components in Economic Time Series”. *Handbook of Applied Econometrics, Volume I: Macroeconomics*. Ed. by M. Hashem Pesaran and Michael R. Wickens. Blackwell.

Orphanides, Athanasios (2001). “Monetary policy rules based on real-time data”. *American Economic Review 81* (4), 964–985.

Pfeifer, Johannes (2013). “An Introduction to Graphs in Dynare”. Tech. rep. University of Mannheim.

Rabanal, Pau and Juan F. Rubio-Ramírez (2005). “Comparing New Keynesian models of the business cycle: A Bayesian approach”. *Journal of Monetary Economics 52*, 1151–1166.

Ratto, Marco and Nikolay Iskrev (2011). "Algorithms for identification analysis under the Dynare environment: final version of the software". Tech. rep. Joint Research Centre, European Commission.

Ruge-Murcia, Francisco J. (2012). "Estimating nonlinear DSGE models by the simulated method of moments: With an application to business cycles". *Journal of Economic Dynamics and Control 36* (6), 914–938.

Sargent, Thomas J (1989). "Two Models of Measurements and the Investment Accelerator". *Journal of Political Economy 97* (2), 251–87.

Schmitt-Grohé, Stephanie and Martín Uribe (2012). "What's news in business cycles". *Econometrica 80* (6), 2733–2764.

Shimer, Robert (2005). "The cyclical behavior of equilibrium unemployment and vacancies". *American Economic Review 95* (1), 25–49.

Shiskin, Julius, Allan H Young, and John C Musgrave (1965). "The X-11 variant of the census method II seasonal adjustment program". Technical Paper 15. US Department of Commerce, Bureau of the Census.

Smets, Frank and Rafael Wouters (2007). "Shocks and frictions in US business cycles: A Bayesian DSGE approach". *American Economic Review 97* (3), 586–606.

Stock, James H. and Mark W. Watson (1999). "Forecasting Inflation". *Journal of Monetary Economics 44* (2), 293–335.

U.S. Census Bureau (2013). "X-13ARIMA-SEATS Reference Manual". Tech. rep. U.S. Census Bureau.

Whelan, Karl (2002). "A Guide To U.S. Chain Aggregated NIPA Data". *Review of Income and Wealth 48* (2), 217–233.

——— (2003). "A Two-Sector Approach to Modeling U.S. NIPA Data". *Journal of Money, Credit and Banking 35* (4), 627–56.

Whittaker, Edmund Taylor (1922). "On a New Method of Graduation". *Proceedings of the Edinburgh Mathematical Society 41*, 63–75.